Doob's Upcrossing Inequality and Martingale Convergence Theorem

Ata Keskin

June 6, 2026

Abstract

In this entry, we formalize Doob's upcrossing inequality and subsequently prove Doob's first martingale convergence theorem. The upcrossing inequality is a fundamental result in the study of martingales. It provides a bound on the expected number of times a submartingale crosses a certain threshold within a given interval. Doob's martingale convergence theorem states that, if we have a submartingale where the supremum over the mean of the positive parts is finite, then the limit process exists almost surely and is integrable. Equivalent statements for martingales and supermartingales are also provided as corollaries.

The proofs provided are based mostly on the formalization done in the Lean mathematical library [1,2].

Contents

1 Introduction

Martingales, in the context of stochastic processes, are encountered in various real-world scenarios where outcomes are influenced by past events but are not entirely predictable due to randomness or uncertainty. A martingale is a stochastic process in which the expected value of the next observation, given all past observations, is equal to the current observation.

One real-world example can be encountered in environmental monitoring, particularly in the study of river flow rates. Consider a hydrologist tasked with monitoring the flow rate of a river to understand its behavior over time. The flow rate of a river is influenced by various factors such as rainfall, snowmelt, groundwater levels, and human activities like dam releases or water diversions. These factors contribute to the variability and unpredictability of the flow rate. In this scenario, the flow rate of the river can be modeled as a martingale. The flow rate at any given time is influenced by past events but is not entirely predictable due to the random nature of rainfall and other factors.

One concept that comes up frequently in the study of martingales are upcrossings and downcrossings. Upcrossings and downcrossings are random variables representing when the value of a stochastic process leaves a fixed interval. Specifically, an upcrossing occurs when the process moves from below the lower bound of the interval to above the upper bound [4], indicating a potential upward trend or positive movement. Conversely, a downcrossing happens when the process crosses below the lower bound of the interval, suggesting a potential downward trend or negative movement. By analyzing the frequency and timing of these crossings, researchers can infer information about the underlying dynamics of the process and detect shifts in its behavior.

For instance, consider tracking the movement of a stock price over time. The process representing the stock's price might cross above a certain threshold (upcrossing) or below it (downcrossing) multiple times during a trading session. The number of such crossings provides insights into the volatility and the trend of the stock.

Doob's upcrossing inequality is a fundamental result in the study of martingales. It provides a bound on the expected number of upcrossings a submartingale undertakes before some point in time.

Let's consider our example concerning river flow rates again. In this context, upcrossings represent instances where the flow rate of the river rises above a certain threshold. For example, the flow rate might cross a threshold indicating flood risk. Downcrossings, on the other hand, represent instances where the flow rate decreases below a certain threshold. This could indicate drought conditions or low-flow periods.

Doob's first martingale convergence theorem gives sufficient conditions for a submartingale to converge to a random variable almost surely. The proof is based on controlling the rate of growth or fluctuations of the submartingale, which is where the *upcrossing inequality* comes into play. By bounding these fluctuations, we can ensure that the submartingale does not exhibit wild behavior or grow too quickly, which is essential for proving convergence.

Formally, the convergence theorem states that, if $(M_n)_{n \geq 0}$ is a submartingale with $\sup_n \mathbb{E}[M_n^+] < \infty$, where M_n^+ denotes the positive part of M_n, then the limit process $M_\infty := \lim_n M_n$ exists almost surely and is integrable. Furthermore, the limit process is measurable with respect to the smallest σ-algebra containing all of the σ-algebras in the filtration. In our formalization, we also show equivalent convergence statements for martingales and supermartingales. The theorem can be used to easily show convergence results for simple scenarios.

Consider the following example: Imagine a casino game where a player bets on the outcome of a random coin toss, where the coin comes up heads with odds $p \in [0, \frac{1}{2})$. Assume that the player goes bust when they have no money remaining. The player's wealth over time can be modeled as a supermartingale, where the value of their wealth at each time step depends only on the outcome of the previous coin toss. Doob's martingale convergence theorem assures us that the player will go bankrupt as the number of coin tosses increases.

The theorem that we have described here and formalized in the scope of our project is called *Doob's first martingale convergence theorem*. It is important to note that the convergence in this theorem is pointwise, not uniform, and is unrelated to convergence in mean square, or indeed in any L^p space. In order to obtain convergence in L^1 (i.e., convergence in mean), one requires uniform integrability of the random variables. In this form, the theorem is called *Doob's second martingale convergence theorem*. Since uniform integrability is not yet formalized in Isabelle/HOL, we have decided to confine our formalization to the first convergence theorem only.

2 Stopping Times and Hitting Times

In this section we formalize stopping times and hitting times. A stopping time is a random variable that represents the time at which a certain event occurs within a stochastic process. A hitting time, also known as first passage time or first hitting time, is a specific type of stopping time that represents the first time a stochastic process reaches a particular state or crosses a certain threshold.

theory *Stopping-Time*
imports *Martingales.Stochastic-Process*
begin

2.1 Stopping Time

The formalization of stopping times here is simply a rewrite of the document *HOL−Probability.Stopping-Time* [5]. We have adapted the document to use the locales defined in our formalization of filtered measure spaces [6] [7]. This way, we can omit the partial formalization of filtrations in the original document. Furthermore, we can include the initial time index t_0 that we introduced as well.

context *linearly-filtered-measure*
begin

— A stopping time is a measurable function from the measure space (possible events) into the time axis.

definition *stopping-time* :: $('a \Rightarrow 'b) \Rightarrow$ *bool* **where**
stopping-time $T = ((T \in \textit{space } M \to \{t_0..\}) \wedge (\forall\, t{\geq}t_0.\ \textit{Measurable.pred } (F\ t)\ (\lambda x.\ T\ x \leq t)))$

lemma *stopping-time-cong*:
assumes $\bigwedge t\ x.\ t \geq t_0 \Longrightarrow x \in \textit{space } (F\ t) \Longrightarrow T\ x = S\ x$
shows *stopping-time* $T =$ *stopping-time* S
proof (*cases* $T \in \textit{space } M \to \{t_0..\}$)
case *True*
hence $S \in \textit{space } M \to \{t_0..\}$ **using** *assms space-F* **by** *force*
then show *?thesis* **unfolding** *stopping-time-def*
using *assms arg-cong*[**where** $f{=}(\lambda P.\ \forall\, t{\geq}t_0.\ P\ t)$] *measurable-cong*[**where** $M{=}F$ - **and** $f{=}\lambda x.\ T\ x \leq$ - **and** $g{=}\lambda x.\ S\ x \leq$ -] *True* **by** *metis*
next
case *False*
hence $S \notin \textit{space } M \to \{t_0..\}$ **using** *assms space-F* **by** *force*
then show *?thesis* **unfolding** *stopping-time-def* **using** *False* **by** *blast*
qed

lemma *stopping-time-ge-zero*:
assumes *stopping-time* T $\omega \in \textit{space } M$

shows $T\ \omega \geq t_0$
using *assms* **unfolding** *stopping-time-def* **by** *auto*

lemma *stopping-timeD*:
assumes *stopping-time* $T\ t \geq t_0$
shows *Measurable.pred* $(F\ t)$ $(\lambda x.\ T\ x \leq t)$
using *assms* **unfolding** *stopping-time-def* **by** *simp*

lemma *stopping-timeI*[*intro?*]:
assumes $\bigwedge x.\ x \in$ *space* $M \Longrightarrow T\ x \geq t_0$
$(\bigwedge t.\ t \geq t_0 \Longrightarrow$ *Measurable.pred* $(F\ t)$ $(\lambda x.\ T\ x \leq t))$
shows *stopping-time* T
using *assms* **by** (*auto simp*: *stopping-time-def*)

lemma *stopping-time-measurable*:
assumes *stopping-time* T
shows $T \in$ *borel-measurable* M
proof (*rule borel-measurableI-le*)
{
fix t **assume** $\neg\ t \geq t_0$
hence $\{x \in$ *space* $M.\ T\ x \leq t\} = \{\}$ **using** *assms dual-order.trans*[*of* - $t\ t_0$] **unfolding** *stopping-time-def* **by** *fastforce*
hence $\{x \in$ *space* $M.\ T\ x \leq t\} \in$ *sets* M **by** (*metis sets.empty-sets*)
}
moreover
{
fix t **assume** *asm*: $t \geq t_0$
hence $\{x \in$ *space* $M.\ T\ x \leq t\} \in$ *sets* M **using** *stopping-timeD*[*OF assms asm*] *sets-F-subset* **unfolding** *Measurable.pred-def space-F*[*OF asm*] **by** *blast*
}
ultimately show $\{x \in$ *space* $M.\ T\ x \leq t\} \in$ *sets* M **for** t **by** *blast*
qed

lemma *stopping-time-const*:
assumes $t \geq t_0$
shows *stopping-time* $(\lambda x.\ t)$ **using** *assms* **by** (*auto simp*: *stopping-time-def*)

lemma *stopping-time-min*:
assumes *stopping-time* T *stopping-time* S
shows *stopping-time* $(\lambda x.\ min\ (T\ x)\ (S\ x))$
using *assms* **by** (*auto simp*: *stopping-time-def min-le-iff-disj intro*!: *pred-intros-logic*)

lemma *stopping-time-max*:
assumes *stopping-time* T *stopping-time* S
shows *stopping-time* $(\lambda x.\ max\ (T\ x)\ (S\ x))$
using *assms* **by** (*auto simp*: *stopping-time-def intro*!: *pred-intros-logic max.coboundedI1*)

2.2 σ-algebra of a Stopping Time

Moving on, we define the σ-algebra associated with a stopping time T. It contains all the information up to time T, the same way $F\ t$ contains all the information up to time t.

definition *pre-sigma* :: ('*a* ⇒ '*b*) ⇒ '*a measure* **where**
pre-sigma T = *sigma* (*space M*) {*A* ∈ *sets M*. ∀ *t*≥t_0. {ω ∈ *A*. *T* ω ≤ *t*} ∈ *sets* (*F t*)}

lemma *measure-pre-sigma*[*simp*]: *emeasure* (*pre-sigma T*) = (λ-. *0*) **by** (*simp add*: *pre-sigma-def emeasure-sigma*)

lemma *sigma-algebra-pre-sigma*:
assumes *stopping-time T*
shows *sigma-algebra* (*space M*) {*A* ∈ *sets M*. ∀ *t*≥t_0. {ω∈*A*. *T* ω ≤ *t*} ∈ *sets* (*F t*)}
proof −
let *?Σ* = {*A* ∈ *sets M*. ∀ *t*≥t_0. {ω∈*A*. *T* ω ≤ *t*} ∈ *sets* (*F t*)}
{
fix *A* **assume** *asm*: *A* ∈ *?Σ*
{
fix *t* **assume** *asm'*: *t* ≥ t_0
hence {ω∈*A*. *T* ω ≤ *t*} ∈ *sets* (*F t*) **using** *asm* **by** *blast*
then have {ω ∈ *space* (*F t*). *T* ω ≤ *t*} − {ω ∈ *A*. *T* ω ≤ *t*} ∈ *sets* (*F t*) **using** *assms*[*THEN stopping-timeD*, *OF asm'*] **by** *auto*
also have {ω ∈ *space* (*F t*). *T* ω ≤ *t*} − {ω ∈ *A*. *T* ω ≤ *t*} = {ω ∈ *space M* − *A*. *T* ω ≤ *t*} **using** *space-F*[*OF asm'*] **by** *blast*
finally have {ω ∈ (*space M*) − *A*. *T* ω ≤ *t*} ∈ *sets* (*F t*) **.**
}
hence *space M* − *A* ∈ *?Σ* **using** *asm* **by** *blast*
}
moreover
{
fix *A* :: *nat* ⇒ '*a set* **assume** *asm*: *range A* ⊆ *?Σ*
{
fix *t* **assume** *t* ≥ t_0
then have (⋃*i*. {ω ∈ *A i*. *T* ω ≤ *t*}) ∈ *sets* (*F t*) **using** *asm* **by** *auto*
also have (⋃*i*. {ω ∈ *A i*. *T* ω ≤ *t*}) = {ω ∈ ⋃(*A* ‘ *UNIV*). *T* ω ≤ *t*} **by** *auto*
finally have {ω ∈ ⋃(*range A*). *T* ω ≤ *t*} ∈ *sets* (*F t*) **.**
}
hence ⋃(*range A*) ∈ *?Σ* **using** *asm* **by** *blast*
}
ultimately show *?thesis* **unfolding** *sigma-algebra-iff2* **by** (*auto intro*!: *sets.sets-into-space*[*THEN PowI*, *THEN subsetI*])
qed

lemma *space-pre-sigma*[*simp*]: *space* (*pre-sigma T*) = *space M* **unfolding** *pre-sigma-def*
by (*intro space-measure-of-conv*)

lemma *sets-pre-sigma*:
assumes *stopping-time T*
shows *sets* (*pre-sigma T*) = $\{A \in$ *sets M*. $\forall\, t \geq t_0.\ \{\omega \in A.\ T\ \omega \leq t\} \in F\ t\}$
unfolding *pre-sigma-def* **using** *sigma-algebra.sets-measure-of-eq*[*OF sigma-algebra-pre-sigma*, *OF assms*] **by** *blast*

lemma *sets-pre-sigmaI*:
assumes *stopping-time T*
and $\bigwedge t.\ t \geq t_0 \Longrightarrow \{\omega \in A.\ T\ \omega \leq t\} \in F\ t$
shows $A \in$ *pre-sigma T*
proof (*cases* $\exists\, t \geq t_0.\ \forall\, t'.\ t' \leq t$)
case *True*
then obtain t **where** $t \geq t_0$ $\{\omega \in A.\ T\ \omega \leq t\} = A$ **by** *blast*
hence $A \in M$ **using** *assms*(*2*)[*of t*] *sets-F-subset*[*of t*] **by** *fastforce*
thus *?thesis* **using** *assms*(*2*) **unfolding** *sets-pre-sigma*[*OF assms*(*1*)] **by** *blast*
next
case *False*
hence $*$: $\{t<..\} \neq \{\}$ **if** $t \geq t_0$ **for** t **by** (*metis not-le empty-iff greaterThan-iff*)
obtain D :: *'b set* **where** D: *countable D* $\bigwedge X.$ *open X* $\Longrightarrow X \neq \{\} \Longrightarrow D \cap X \neq \{\}$ **by** (*metis countable-dense-setE disjoint-iff*)
have $**$: $D \cap \{t<..\} \neq \{\}$ **if** $t \geq t_0$ **for** t **using** $*$ *that* **by** (*intro D*(*2*)) *auto*
{
fix ω
obtain t **where** t: $t \geq t_0$ $T\ \omega \leq t$ **using** *linorder-linear* **by** *auto*
moreover obtain t' **where** $t' \in D \cap \{t<..\} \cap \{t_0..\}$ **using** $**$ t **by** *fastforce*
moreover have $T\ \omega \leq t'$ **using** *calculation* **by** *fastforce*
ultimately have $\exists\, t.\ \exists\, t' \in D \cap \{t<..\} \cap \{t_0..\}.\ T\ \omega \leq t'$ **by** *blast*
}
hence $(\bigcup t' \in (\bigcup t.\ D \cap \{t<..\} \cap \{t_0..\}).\ \{\omega \in A.\ T\ \omega \leq t'\}) = A$ **by** *fast*
moreover have $(\bigcup t' \in (\bigcup t.\ D \cap \{t<..\} \cap \{t_0..\}).\ \{\omega \in A.\ T\ \omega \leq t'\}) \in M$ **using** D *assms*(*2*) *sets-F-subset* **by** (*intro sets.countable-UN''*, *fastforce*, *fast*)
ultimately have $A \in M$ **by** *argo*
thus *?thesis* **using** *assms*(*2*) **unfolding** *sets-pre-sigma*[*OF assms*(*1*)] **by** *blast*
qed

lemma *pred-pre-sigmaI*:
assumes *stopping-time T*
shows $(\bigwedge t.\ t \geq t_0 \Longrightarrow$ *Measurable.pred* $(F\ t)$ $(\lambda\omega.\ P\ \omega \wedge T\ \omega \leq t)) \Longrightarrow$ *Measurable.pred* (*pre-sigma T*) P
unfolding *pred-def space-pre-sigma* **using** *assms* **by** (*auto intro*: *sets-pre-sigmaI*[*OF assms*(*1*)])

lemma *sets-pre-sigmaD*:
assumes *stopping-time T* $A \in$ *pre-sigma T* $t \geq t_0$
shows $\{\omega \in A.\ T\ \omega \leq t\} \in$ *sets* $(F\ t)$
using *assms sets-pre-sigma* **by** *auto*

lemma *borel-measurable-stopping-time-pre-sigma*:

assumes *stopping-time T*
shows $T \in$ *borel-measurable* (*pre-sigma T*)
proof (*intro borel-measurableI-le sets-pre-sigmaI*[*OF assms*])
fix t t' **assume** *asm*: $t \geq t_0$
{
assume $\neg\ t' \geq t_0$
hence $\{\omega \in \{x \in$ *space* (*pre-sigma T*). $T\ x \leq t'\}$. $T\ \omega \leq t\} = \{\}$ **using** *assms dual-order.trans*[*of - t' t_0*] **unfolding** *stopping-time-def* **by** *fastforce*
hence $\{\omega \in \{x \in$ *space* (*pre-sigma T*). $T\ x \leq t'\}$. $T\ \omega \leq t\} \in$ *sets* (F t) **by** (*metis sets.empty-sets*)
}
moreover
{
assume *asm'*: $t' \geq t_0$
have $\{\omega \in$ *space* (F (*min* t' t)). $T\ \omega \leq$ *min* t' $t\} \in$ *sets* (F (*min* t' t))
using *assms asm asm'* **unfolding** *pred-def*[*symmetric*] **by** (*intro stopping-timeD*) *auto*
also have ... $\subseteq$ *sets* (F t)
using *assms asm asm'* **by** (*intro sets-F-mono*) *auto*
finally have $\{\omega \in \{x \in$ *space* (*pre-sigma T*). $T\ x \leq t'\}$. $T\ \omega \leq t\} \in$ *sets* (F t)
using *asm asm'* **by** *simp*
}
ultimately show $\{\omega \in \{x \in$ *space* (*pre-sigma T*). $T\ x \leq t'\}$. $T\ \omega \leq t\} \in$ *sets* (F t) **by** *blast*
qed

lemma *mono-pre-sigma*:
assumes *stopping-time T stopping-time S*
and $\bigwedge x.\ x \in$ *space M* $\Longrightarrow T\ x \leq S\ x$
shows *pre-sigma T* $\subseteq$ *pre-sigma S*
proof
fix A **assume** $A \in$ *pre-sigma T*
hence *asm*: $A \in$ *sets M* $t \geq t_0 \Longrightarrow \{\omega \in A.\ T\ \omega \leq t\} \in$ *sets* (F t) **for** t **using** *assms sets-pre-sigma* **by** *blast+*
{
fix t **assume** *asm'*: $t \geq t_0$
then have $A \subseteq$ *space M* **using** *sets.sets-into-space asm* **by** *blast*
have $\{\omega{\in}A.\ T\ \omega \leq t\} \cap \{\omega{\in}$*space* ($F$ t). $S\ \omega \leq t\} \in$ *sets* (F t)
using *asm asm' stopping-timeD*[*OF assms*(*2*)] **by** (*auto simp*: *pred-def*)
also have $\{\omega{\in}A.\ T\ \omega \leq t\} \cap \{\omega{\in}$*space* ($F$ t). $S\ \omega \leq t\} = \{\omega{\in}A.\ S\ \omega \leq t\}$
using *sets.sets-into-space*[*OF asm*(*1*)] *assms*(*3*) *order-trans asm'* **by** *fastforce*
finally have $\{\omega{\in}A.\ S\ \omega \leq t\} \in$ *sets* (F t) **by** *simp*
}
thus $A \in$ *pre-sigma S* **by** (*intro sets-pre-sigmaI assms asm*) *blast*
qed

lemma *stopping-time-measurable-le*:
assumes *stopping-time T* $s \geq t_0$ $t \geq s$
shows *Measurable.pred* (F t) ($\lambda\omega.\ T\ \omega \leq s$)

using *assms stopping-timeD*[*of T*] *sets-F-mono*[*of - t*] **by** (*auto simp*: *pred-def*)

lemma *stopping-time-measurable-less*:
assumes *stopping-time T s* $\geq$ t_0 *t* $\geq$ *s*
shows *Measurable.pred* (*F t*) ($\lambda\omega$. *T* ω < *s*)
proof −
have *Measurable.pred* (*F t*) ($\lambda\omega$. *T* ω < *t*) **if** *asm*: *stopping-time T t* $\geq$ t_0 **for** *T t*
proof −
obtain *D* :: *'b set* **where** *D*: *countable D* $\bigwedge X$. *open X* $\Longrightarrow$ *X* $\neq$ {} $\Longrightarrow$ *D* $\cap$ *X* $\neq$ {} **by** (*metis countable-dense-setE disjoint-iff*)
show *?thesis*
proof *cases*
assume $*$: $\forall t'\in\{t_0..<t\}$. $\{t'<..<t\} \neq \{\}$
hence $**$: *D* $\cap$ $\{t'<..< t\} \neq \{\}$ **if** $t' \in \{t_0..<t\}$ **for** *t'* **using** *that* **by** (*intro D*(*2*)) *fastforce+*

show *?thesis*
proof (*rule measurable-cong*[*THEN iffD2*])
show *T* ω < *t* $\longleftrightarrow$ ($\exists r\in D \cap \{t_0..<t\}$. *T* $\omega \leq r$) **if** $\omega \in$ *space* (*F t*) **for** ω
using $**$[*of T* ω] *that less-imp-le stopping-time-ge-zero asm* **by** *fastforce*
show *Measurable.pred* (*F t*) (λw. $\exists r\in D \cap \{t_0..<t\}$. *T w* $\leq$ *r*)
using *stopping-time-measurable-le asm D* **by** (*intro measurable-pred-countable*) *auto*
qed
next
assume $\neg$ ($\forall t'\in\{t_0..<t\}$. $\{t'<..<t\} \neq \{\}$)
then obtain *t'* **where** *t'*: $t' \in \{t_0..<t\}$ $\{t'<..<t\} = \{\}$ **by** *blast*
show *?thesis*
proof (*rule measurable-cong*[*THEN iffD2*])
show *T* ω < *t* $\longleftrightarrow$ *T* $\omega \leq t'$ **for** ω **using** *t'* **by** (*metis atLeastLessThan-iff emptyE greaterThanLessThan-iff linorder-not-less order.strict-trans1*)
show *Measurable.pred* (*F t*) (λw. *T w* $\leq$ *t'*) **using** *t'* **by** (*intro stopping-time-measurable-le*[*OF asm*(*1*)]) *auto*
qed
qed
qed
thus *?thesis*
using *assms sets-F-mono*[*of - t*] **by** (*auto simp add*: *pred-def*)
qed

lemma *stopping-time-measurable-ge*:
assumes *stopping-time T s* $\geq$ t_0 *t* $\geq$ *s*
shows *Measurable.pred* (*F t*) ($\lambda\omega$. *T* $\omega \geq s$)
by (*auto simp*: *not-less*[*symmetric*] *intro*: *stopping-time-measurable-less*[*OF assms*] *Measurable.pred-intros-logic*)

lemma *stopping-time-measurable-gr*:
assumes *stopping-time T s* $\geq$ t_0 *t* $\geq$ *s*
shows *Measurable.pred* (*F t*) (λx. *s* < *T x*)

by (*auto simp add*: *not-le*[*symmetric*] *intro*: *stopping-time-measurable-le*[*OF assms*] *Measurable.pred-intros-logic*)

lemma *stopping-time-measurable-eq*:
assumes *stopping-time* T $s \geq t_0$ $t \geq s$
shows *Measurable.pred* (F t) ($\lambda\omega$. T ω = s)
unfolding *eq-iff* **using** *stopping-time-measurable-le*[*OF assms*] *stopping-time-measurable-ge*[*OF assms*]
by (*intro pred-intros-logic*)

lemma *stopping-time-less-stopping-time*:
assumes *stopping-time* T *stopping-time* S
shows *Measurable.pred* (*pre-sigma* T) ($\lambda\omega$. T ω < S ω)
proof (*rule pred-pre-sigmaI*)
fix t **assume** *asm*: $t \geq t_0$
obtain D :: *'b set* **where** D: *countable* D **and** *semidense-D*: $\bigwedge x\ y.\ x < y \Longrightarrow (\exists\, b \in D.\ x \leq b \wedge b < y)$
using *countable-separating-set-linorder2* **by** *auto*
show *Measurable.pred* (F t) ($\lambda\omega$. $T\ \omega < S\ \omega \wedge T\ \omega \leq t$)
proof (*rule measurable-cong*[*THEN iffD2*])
let *?f* = $\lambda\omega$. *if* $T\ \omega = t$ *then* $\neg$ $S\ \omega \leq t$ *else* $\exists\, s \in D \cap \{t_0..t\}.\ T\ \omega \leq s \wedge \neg (S\ \omega \leq s)$
{
fix ω **assume** $\omega \in$ *space* (F t) $T\ \omega \leq t$ $T\ \omega \neq t$ $T\ \omega < S\ \omega$
hence $t_0 \leq T\ \omega$ $T\ \omega <$ *min* t (S ω) **using** *asm stopping-time-ge-zero*[*OF assms*(*1*)] **by** *auto*
then obtain r **where** $r \in D$ $t_0 \leq r$ $T\ \omega \leq r$ $r <$ *min* t (S ω) **using** *semidense-D order-trans* **by** *blast*
hence $\exists\, s \in D \cap \{t_0..t\}.\ T\ \omega \leq s \wedge s < S\ \omega$ **by** *auto*
}
thus ($T\ \omega < S\ \omega \wedge T\ \omega \leq t$) = *?f* ω **if** $\omega \in$ *space* (F t) **for** ω **using** *that* **by** *force*
show *Measurable.pred* (F t) *?f*
using *assms asm D*
by (*intro pred-intros-logic measurable-If measurable-pred-countable countable-Collect*
stopping-time-measurable-le predE stopping-time-measurable-eq) *auto*
qed
qed (*intro assms*)

end

lemma (**in** *enat-filtered-measure*) *stopping-time-SUP-enat*:
fixes T :: *nat* $\Rightarrow$ (*'a* $\Rightarrow$ *enat*)
shows ($\bigwedge i$. *stopping-time* (T i)) $\Longrightarrow$ *stopping-time* (*SUP* i. T i)
unfolding *stopping-time-def SUP-apply SUP-le-iff* **by** (*auto intro*!: *pred-intros-countable*)

lemma (**in** *enat-filtered-measure*) *stopping-time-Inf-enat*:
assumes $\bigwedge i$. *Measurable.pred* (F i) (P i)

shows *stopping-time* (λω. *Inf* {*i*. *P i* ω})
proof −
{
fix *t* :: *enat* **assume** *asm*: *t* ≠ ∞
moreover
{
fix *i* ω **assume** *Inf* {*i*. *P i* ω} ≤ *t*
moreover have *a* < *eSuc b* ⟷ (*a* ≤ *b* ∧ *a* ≠ ∞) **for** *a b* **by** (*cases a*) *auto*
ultimately have (∃ *i*≤*t*. *P i* ω) **using** *asm* **unfolding** *Inf-le-iff* **by** (*cases t*) (*auto elim*!: *allE*[*of - eSuc t*])
}
ultimately have ∗: ⋀ω. *Inf* {*i*. *P i* ω} ≤ *t* ⟷ (∃ *i*∈{..*t*}. *P i* ω) **by** (*auto intro*!: *Inf-lower2*)
have *Measurable.pred* (*F t*) (λω. *Inf* {*i*. *P i* ω} ≤ *t*) **unfolding** ∗ **using** *sets-F-mono assms* **by** (*intro pred-intros-countable-bounded*) (*auto simp*: *pred-def*)
}
moreover have *Measurable.pred* (*F t*) (λω. *Inf* {*i*. *P i* ω} ≤ *t*) **if** *t* = ∞ **for** *t* **using** *that* **by** *simp*
ultimately show *?thesis* **by** (*blast intro*: *stopping-timeI*[*OF i0-lb*])
qed

lemma (**in** *nat-filtered-measure*) *stopping-time-Inf-nat*:
assumes ⋀*i*. *Measurable.pred* (*F i*) (*P i*)
⋀*i* ω. ω ∈ *space M* ⟹ ∃ *n*. *P n* ω
shows *stopping-time* (λω. *Inf* {*i*. *P i* ω})
proof (*rule stopping-time-cong*[*THEN iffD2*])
show *stopping-time* (λ*x*. *LEAST n*. *P n x*)
proof
fix *t*
have ((*LEAST n*. *P n* ω) ≤ *t*) = (∃ *i*≤*t*. *P i* ω) **if** ω ∈ *space M* **for** ω **by** (*rule LeastI2-wellorder-ex*[*OF assms*(*2*)[*OF that*]]) *auto*
moreover have *Measurable.pred* (*F t*) (λ*w*. ∃ *i*∈{..*t*}. *P i w*) **using** *sets-F-mono*[*of - t*] *assms* **by** (*intro pred-intros-countable-bounded*) (*auto simp*: *pred-def*)
ultimately show *Measurable.pred* (*F t*) (λω. (*LEAST n*. *P n* ω) ≤ *t*) **by** (*subst measurable-cong*[*of F t*]) *auto*
qed (*simp*)
qed (*simp add*: *Inf-nat-def*)

definition *stopped-value* :: (′*b* ⇒ ′*a* ⇒ ′*c*) ⇒ (′*a* ⇒ ′*b*) ⇒ (′*a* ⇒ ′*c*) **where**
stopped-value X τ ω = *X* (τ ω) ω

2.3 Hitting Time

Given a stochastic process X and a borel set A, *hitting-time X A s t* is the first time X is in A after time s and before time t. If X does not hit A after time s and before t then the hitting time is simply t. The definition presented here coincides with the definition of hitting times in mathlib [1].

context *linearly-filtered-measure*

begin

definition *hitting-time* :: $('b \Rightarrow 'a \Rightarrow 'c) \Rightarrow 'c\ set \Rightarrow 'b \Rightarrow 'b \Rightarrow ('a \Rightarrow 'b)$ **where**
hitting-time $X\ A\ s\ t = (\lambda\omega.\ if\ \exists\, i{\in}\{s..t\} \cap \{t_0..\}.\ X\ i\ \omega \in A\ then\ Inf\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\})\ else\ max\ t_0\ t)$

lemma *hitting-time-def'*:
hitting-time $X\ A\ s\ t = (\lambda\omega.\ Inf\ (insert\ (max\ t_0\ t)\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\})))$
proof *cases*
assume *asm*: $t_0 \le s \wedge s \le t$
hence $\{s..t\} \cap \{t_0..\} = \{s..t\}$ **by** *simp*
{
fix ω
assume $*$: $\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\} \neq \{\}$
then obtain i **where** $i \in \{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\}$ **by** *blast*
hence $Inf\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\}) \le t$ **by** (*intro cInf-lower*[*of i*, *THEN order-trans*]) *auto*
hence $Inf\ (insert\ (max\ t_0\ t)\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\})) = Inf\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\})$ **using** *asm* $*$ *inf-absorb2* **by** (*subst cInf-insert-If*) *force+*
also have ... = *hitting-time* $X\ A\ s\ t\ \omega$ **using** $*$ **unfolding** *hitting-time-def* **by** *auto*
finally have *hitting-time* $X\ A\ s\ t\ \omega = Inf\ (insert\ (max\ t_0\ t)\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\}))$ **by** *argo*
}
moreover
{
fix ω
assume $\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\} = \{\}$
hence *hitting-time* $X\ A\ s\ t\ \omega = Inf\ (insert\ (max\ t_0\ t)\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\}))$ **unfolding** *hitting-time-def* **by** *auto*
}
ultimately show *?thesis* **by** *fast*
next
assume $\neg\ (t_0 \le s \wedge s \le t)$
moreover
{
assume *asm*: $s < t_0\ t \ge t_0$
hence $\{s..t\} \cap \{t_0..\} = \{t_0..t\}$ **by** *simp*
{
fix ω
assume $*$: $\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\} \neq \{\}$
then obtain i **where** $i \in \{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\}$ **by** *blast*
hence $Inf\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\}) \le t$ **by** (*intro cInf-lower*[*of i*, *THEN order-trans*]) *auto*
hence $Inf\ (insert\ (max\ t_0\ t)\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\})) = Inf\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\})$ **using** *asm* $*$ *inf-absorb2* **by** (*subst cInf-insert-If*) *force+*

also have ... = *hitting-time* $X\ A\ s\ t\ \omega$ **using** $*$ **unfolding** *hitting-time-def*

by *auto*
finally have *hitting-time* $X\ A\ s\ t\ \omega = Inf\ (insert\ (max\ t_0\ t)\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\}))$ **by** *argo*
}
moreover
{
fix ω
assume $\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\} = \{\}$
hence *hitting-time* $X\ A\ s\ t\ \omega = Inf\ (insert\ (max\ t_0\ t)\ (\{s..t\} \cap \{t_0..\} \cap \{i.\ X\ i\ \omega \in A\}))$ **unfolding** *hitting-time-def* **by** *auto*
}
ultimately have *?thesis* **by** *fast*
}
moreover have *?thesis* **if** $s < t_0\ t < t_0$ **using** *that* **unfolding** *hitting-time-def* **by** *auto*
moreover have *?thesis* **if** $s > t$ **using** *that* **unfolding** *hitting-time-def* **by** *auto*
ultimately show *?thesis* **by** *fastforce*
qed

— The following lemma provides a sufficient condition for an injective function to preserve a hitting time.

lemma *hitting-time-inj-on*:
assumes *inj-on f S* $\bigwedge\omega\ t.\ t \geq t_0 \Longrightarrow X\ t\ \omega \in S\ A \subseteq S$
shows *hitting-time* $X\ A =$ *hitting-time* $(\lambda t\ \omega.\ f\ (X\ t\ \omega))\ (f\ `\ A)$
proof −
have $X\ t\ \omega \in A \longleftrightarrow f\ (X\ t\ \omega) \in f\ `\ A$ **if** $t \geq t_0$ **for** $t\ \omega$ **using** *assms that inj-on-image-mem-iff* **by** *meson*
hence $\{t_0..\} \cap \{i.\ X\ i\ \omega \in A\} = \{t_0..\} \cap \{i.\ f\ (X\ i\ \omega) \in f\ `\ A\}$ **for** ω **by** *blast*
thus *?thesis* **unfolding** *hitting-time-def′ Int-assoc* **by** *presburger*
qed

lemma *hitting-time-translate*:
fixes *c* :: - :: *ab-group-add*
shows *hitting-time* $X\ A =$ *hitting-time* $(\lambda n\ \omega.\ X\ n\ \omega + c)\ (((+)\ c)\ `\ A)$
by (*subst hitting-time-inj-on*[*OF inj-on-add*, *of - UNIV*]) (*simp add*: *add.commute*)+

lemma *hitting-time-le*:
assumes $t \geq t_0$
shows *hitting-time* $X\ A\ s\ t\ \omega \leq t$
unfolding *hitting-time-def′* **using** *assms*
by (*intro cInf-lower*[*of max* t_0 *t*, *THEN order-trans*]) *auto*

lemma *hitting-time-ge*:
assumes $t \geq t_0\ s \leq t$
shows $s \leq$ *hitting-time* $X\ A\ s\ t\ \omega$
unfolding *hitting-time-def′* **using** *assms*
by (*intro le-cInf-iff*[*THEN iffD2*]) *auto*

lemma *hitting-time-mono*:
assumes $t \geq t_0$ $s \leq s'$ $t \leq t'$
shows *hitting-time* X A s t $\omega \leq$ *hitting-time* X A s' t' ω
unfolding *hitting-time-def'* **using** *assms* **by** (*fastforce intro*!: *cInf-mono*)

end

context *nat-filtered-measure*
begin

— Hitting times are stopping times for adapted processes.

lemma *stopping-time-hitting-time*:
assumes *adapted-process* M F 0 X $A \in$ *borel*
shows *stopping-time* (*hitting-time* X A s t)
proof −
interpret *adapted-process* M F 0 X **by** (*rule assms*)
have *insert* t $(\{s..t\} \cap \{i.\ X\ i\ \omega \in A\}) = \{i.\ i = t \vee i \in (\{s..t\} \cap \{i.\ X\ i\ \omega \in A\})\}$ **for** ω **by** *blast*
hence *hitting-time* X A s t $= (\lambda\omega.\ Inf\ \{i.\ i = t \vee i \in (\{s..t\} \cap \{i.\ X\ i\ \omega \in A\})\})$ **unfolding** *hitting-time-def'* **by** *simp*
thus *?thesis* **using** *assms* **by** (*auto intro*: *stopping-time-Inf-nat*)
qed

lemma *stopping-time-hitting-time'*:
assumes *adapted-process* M F 0 X $A \in$ *borel* *stopping-time* s $\bigwedge\omega.\ s\ \omega \leq t$
shows *stopping-time* $(\lambda\omega.$ *hitting-time* X A $(s\ \omega)$ t $\omega)$
proof −
interpret *adapted-process* M F 0 X **by** (*rule assms*)
{
fix n
have $s\ \omega \leq$ *hitting-time* X A $(s\ \omega)$ t ω **if** $s\ \omega > n$ **for** ω **using** *hitting-time-ge*[*OF* - *assms*(*4*)] **by** *simp*
hence $(\bigcup i{\in}\{n{<}..\}.\ \{\omega.\ s\ \omega = i\} \cap \{\omega.$ *hitting-time* $X\ A\ i\ t\ \omega \leq n\}) = \{\}$ **by** *fastforce*
hence $*$: $(\lambda\omega.$ *hitting-time* X A $(s\ \omega)$ t $\omega \leq n) = (\lambda\omega.\ \exists i{\leq}n.\ s\ \omega = i \wedge$ *hitting-time* $X\ A\ i\ t\ \omega \leq n)$ **by** *force*

have *Measurable.pred* $(F\ n)$ $(\lambda\omega.\ s\ \omega = i \wedge$ *hitting-time* $X\ A\ i\ t\ \omega \leq n)$ **if** $i \leq n$ **for** i
proof −
have *Measurable.pred* $(F\ i)$ $(\lambda\omega.\ s\ \omega = i)$ **using** *stopping-time-measurable-eq assms* **by** *blast*
hence *Measurable.pred* $(F\ n)$ $(\lambda\omega.\ s\ \omega = i)$ **by** (*meson less-eq-nat.simps measurable-from-subalg subalgebra-F that*)
moreover have *Measurable.pred* $(F\ n)$ $(\lambda\omega.$ *hitting-time* $X\ A\ i\ t\ \omega \leq n)$ **using** *stopping-timeD*[*OF stopping-time-hitting-time*, *OF assms*(*1*,*2*)] **by** *blast*
ultimately show *?thesis* **by** *auto*
qed

hence *Measurable.pred* $(F\ n)$ $(\lambda\omega.\ \exists i{\le}n.\ s\ \omega = i \wedge$ *hitting-time* $X\ A\ i\ t\ \omega \le n)$ **by** (*intro pred-intros-countable*) *auto*
hence *Measurable.pred* $(F\ n)$ $(\lambda\omega.$ *hitting-time* $X\ A\ (s\ \omega)\ t\ \omega \le n)$ **using** $*$ **by** *argo*
}
thus *?thesis* **by** (*intro stopping-timeI*) *auto*
qed

— If X hits A at time $j \in \{s..t\}$, then the stopped value of X at the hitting time of A in the interval $\{s..t\}$ is an element of A.

lemma *stopped-value-hitting-time-mem*:
assumes $j \in \{s..t\}$ $X\ j\ \omega \in A$
shows *stopped-value* X (*hitting-time* $X\ A\ s\ t$) $\omega \in A$
proof $-$
have $\exists i{\in}\{s..t\} \cap \{0..\}.\ X\ i\ \omega \in A$ **using** *assms* **by** *blast*
moreover have *Inf* $(\{s..t\} \cap \{i.\ X\ i\ \omega \in A\}) \in \{s..t\} \cap \{i.\ X\ i\ \omega \in A\}$ **using** *assms* **by** (*blast intro*!: *Inf-nat-def1*)
ultimately show *?thesis* **unfolding** *hitting-time-def stopped-value-def* **by** *simp*
qed

lemma *hitting-time-le-iff*:
assumes $i < t$
shows *hitting-time* $X\ A\ s\ t\ \omega \le i \longleftrightarrow (\exists j \in \{s..i\}.\ X\ j\ \omega \in A)$ (**is** *?lhs* = *?rhs*)
proof
assume *?lhs*
moreover have *hitting-time* $X\ A\ s\ t\ \omega \in$ *insert* t $(\{s..t\} \cap \{i.\ X\ i\ \omega \in A\})$ **by** (*metis hitting-time-def′ Int-atLeastAtMostR2 inf-sup-aci*(*1*) *insertI1 max-0L wellorder-InfI*)
ultimately have *hitting-time* $X\ A\ s\ t\ \omega \in \{s..i\} \cap \{i.\ X\ i\ \omega \in A\}$ **using** *assms* **by** *force*
thus *?rhs* **by** *blast*
next
assume *?rhs*
then obtain j **where** j: $j \in \{s..i\}$ $X\ j\ \omega \in A$ **by** *blast*
hence *hitting-time* $X\ A\ s\ t\ \omega \le j$ **unfolding** *hitting-time-def′* **using** *assms* **by** (*auto intro*: *cInf-lower*)
thus *?lhs* **using** j **by** *simp*
qed

lemma *hitting-time-less-iff*:
assumes $i \le t$
shows *hitting-time* $X\ A\ s\ t\ \omega < i \longleftrightarrow (\exists j \in \{s..{<}i\}.\ X\ j\ \omega \in A)$ (**is** *?lhs* = *?rhs*)
proof
assume *?lhs*
moreover have *hitting-time* $X\ A\ s\ t\ \omega \in$ *insert* t $(\{s..t\} \cap \{i.\ X\ i\ \omega \in A\})$ **by** (*metis hitting-time-def′ Int-atLeastAtMostR2 inf-sup-aci*(*1*) *insertI1 max-0L wellorder-InfI*)

ultimately have *hitting-time* $X\ A\ s\ t\ \omega \in \{s..<i\} \cap \{i.\ X\ i\ \omega \in A\}$ **using** *assms* **by** *force*
thus *?rhs* **by** *blast*
next
assume *?rhs*
then obtain j **where** j: $j \in \{s..<i\}$ $X\ j\ \omega \in A$ **by** *blast*
hence *hitting-time* $X\ A\ s\ t\ \omega \leq j$ **unfolding** *hitting-time-def'* **using** *assms* **by** (*auto intro*: *cInf-lower*)
thus *?lhs* **using** j **by** *simp*
qed

— If X already hits A in the interval $\{s..t\}$, then *hitting-time* $X\ A\ s\ t =$ *hitting-time* $X\ A\ s\ t'$ for $t \leq t'$.

lemma *hitting-time-eq-hitting-time*:
assumes $t \leq t'$ $j \in \{s..t\}$ $X\ j\ \omega \in A$
shows *hitting-time* $X\ A\ s\ t\ \omega =$ *hitting-time* $X\ A\ s\ t'\ \omega$ (**is** *?lhs* = *?rhs*)
proof −
have *hitting-time* $X\ A\ s\ t\ \omega \in \{s..t'\}$ **using** *hitting-time-le*[*THEN order-trans*, *of* $t\ t'\ X\ A\ s$] *hitting-time-ge*[*of* $t\ s\ X\ A$] *assms* **by** *auto*
moreover have *stopped-value* X (*hitting-time* $X\ A\ s\ t$) $\omega \in A$ **by** (*blast intro*: *stopped-value-hitting-time-mem assms*)
ultimately have *hitting-time* $X\ A\ s\ t'\ \omega \leq$ *hitting-time* $X\ A\ s\ t\ \omega$ **by** (*fastforce simp add*: *hitting-time-def'*[**where** $t{=}t'$] *stopped-value-def intro*!: *cInf-lower*)
thus *?thesis* **by** (*blast intro*: *le-antisym hitting-time-mono*[*OF - order-refl assms*(*1*)])
qed

end

end

3 Doob's Upcrossing Inequality and Martingale Convergence Theorems

In this section we formalize upcrossings and downcrossings. Following this, we prove Doob's upcrossing inequality and first martingale convergence theorem.

theory *Upcrossing*
imports *Martingales.Martingale Stopping-Time*
begin

lemma *real-embedding-borel-measurable*: *real* $\in$ *borel-measurable borel* **by** (*auto intro*: *borel-measurable-continuous-onI*)

lemma *limsup-lower-bound*:

fixes *u*:: *nat* ⇒ *ereal*
assumes *limsup u* > *l*
shows ∃ *N*>*k*. *u N* > *l*
proof −
have *limsup u* = − *liminf* (λ*n*. − *u n*) **using** *liminf-ereal-cminus*[*of 0 u*] **by** *simp*
hence *liminf* (λ*n*. − *u n*) < − *l* **using** *assms ereal-less-uminus-reorder* **by** *presburger*
hence ∃ *N*>*k*. − *u N* < − *l* **using** *liminf-upper-bound* **by** *blast*
thus *?thesis* **using** *ereal-less-uminus-reorder* **by** *simp*
qed

lemma *ereal-abs-max-min*: |*c*| = *max 0 c* − *min 0 c* **for** *c* :: *ereal*
by (*cases c* ≥ *0*) *auto*

3.1 Upcrossings and Downcrossings

Given a stochastic process X, real values a and b, and some point in time N, we would like to define a notion of "upcrossings" of X across the band $\{a..b\}$ which counts the number of times any realization of X crosses from below a to above b before time N. To make this heuristic rigorous, we inductively define the following hitting times.

context *nat-filtered-measure*
begin

context
fixes *X* :: *nat* ⇒ ′*a* ⇒ *real*
and *a b* :: *real*
and *N* :: *nat*
begin

primrec *upcrossing* :: *nat* ⇒ ′*a* ⇒ *nat* **where**
upcrossing 0 = (λω. *0*) |
upcrossing (*Suc n*) = (λω. *hitting-time X* {*b*..} (*hitting-time X* {..*a*} (*upcrossing n* ω) *N* ω) *N* ω)

definition *downcrossing* :: *nat* ⇒ ′*a* ⇒ *nat* **where**
downcrossing n = (λω. *hitting-time X* {..*a*} (*upcrossing n* ω) *N* ω)

lemma *upcrossing-simps*:
upcrossing 0 = (λω. *0*)
upcrossing (*Suc n*) = (λω. *hitting-time X* {*b*..} (*downcrossing n* ω) *N* ω)
by (*auto simp add*: *downcrossing-def*)

lemma *downcrossing-simps*:
downcrossing 0 = *hitting-time X* {..*a*} *0 N*
downcrossing n = (λω. *hitting-time X* {..*a*} (*upcrossing n* ω) *N* ω)
by (*auto simp add*: *downcrossing-def*)

declare *upcrossing.simps*[*simp del*]

lemma *upcrossing-le*: *upcrossing n ω ≤ N*
by (*cases n*) (*auto simp add*: *upcrossing-simps hitting-time-le*)

lemma *downcrossing-le*: *downcrossing n ω ≤ N*
by (*cases n*) (*auto simp add*: *downcrossing-simps hitting-time-le*)

lemma *upcrossing-le-downcrossing*: *upcrossing n ω ≤ downcrossing n ω*
unfolding *downcrossing-simps* **by** (*auto intro*: *hitting-time-ge upcrossing-le*)

lemma *downcrossing-le-upcrossing-Suc*: *downcrossing n ω ≤ upcrossing* (*Suc n*) *ω*
unfolding *upcrossing-simps* **by** (*auto intro*: *hitting-time-ge downcrossing-le*)

lemma *upcrossing-mono*:
assumes *n ≤ m*
shows *upcrossing n ω ≤ upcrossing m ω*
using *order-trans*[*OF upcrossing-le-downcrossing downcrossing-le-upcrossing-Suc*] *assms*
by (*rule lift-Suc-mono-le*)

lemma *downcrossing-mono*:
assumes *n ≤ m*
shows *downcrossing n ω ≤ downcrossing m ω*
using *order-trans*[*OF downcrossing-le-upcrossing-Suc upcrossing-le-downcrossing*] *assms*
by (*rule lift-Suc-mono-le*)

— The following lemmas help us make statements about when an upcrossing (resp. downcrossing) occurs, and the value that the process takes at that instant.

lemma *stopped-value-upcrossing*:
assumes *upcrossing* (*Suc n*) *ω ≠ N*
shows *stopped-value X* (*upcrossing* (*Suc n*)) *ω ≥ b*
proof −
have ∗: *upcrossing* (*Suc n*) *ω < N* **using** *le-neq-implies-less upcrossing-le assms* **by** *presburger*
have ∃ *j* ∈ {*downcrossing n ω..upcrossing* (*Suc n*) *ω*}. *X j ω* ∈ {*b..*}
using *hitting-time-le-iff* [*THEN iffD1*, *OF* ∗] *upcrossing-simps* **by** *fastforce*
then obtain *j* **where** *j*: *j* ∈ {*downcrossing n ω..N*} *X j ω* ∈ {*b..*} **using** ∗ **by** (*meson atLeastatMost-subset-iff le-refl subsetD upcrossing-le*)
thus *?thesis* **using** *stopped-value-hitting-time-mem*[*of j - - X*] **unfolding** *upcrossing-simps stopped-value-def* **by** *blast*
qed

lemma *stopped-value-downcrossing*:
assumes *downcrossing n ω ≠ N*
shows *stopped-value X* (*downcrossing n*) *ω ≤ a*
proof −

have ∗: *downcrossing n ω* < *N* **using** *le-neq-implies-less downcrossing-le assms* **by** *presburger*
have ∃ *j* ∈ {*upcrossing n ω*..*downcrossing n ω*}. *X j ω* ∈ {..*a*}
using *hitting-time-le-iff*[*THEN iffD1*, *OF* ∗] *downcrossing-simps* **by** *fastforce*
then obtain *j* **where** *j*: *j* ∈ {*upcrossing n ω*..*N*} *X j ω* ∈ {..*a*} **using** ∗ **by** (*meson atLeastatMost-subset-iff le-refl subsetD downcrossing-le*)
thus *?thesis* **using** *stopped-value-hitting-time-mem*[*of j - - X*] **unfolding** *downcrossing-simps stopped-value-def* **by** *blast*
qed

lemma *upcrossing-less-downcrossing*:
assumes *a* < *b downcrossing* (*Suc n*) *ω* ≠ *N*
shows *upcrossing* (*Suc n*) *ω* < *downcrossing* (*Suc n*) *ω*
proof −
have *upcrossing* (*Suc n*) *ω* ≠ *N* **using** *assms* **by** (*metis le-antisym downcrossing-le upcrossing-le-downcrossing*)
hence *stopped-value X* (*downcrossing* (*Suc n*)) *ω* < *stopped-value X* (*upcrossing* (*Suc n*)) *ω*
using *assms stopped-value-downcrossing stopped-value-upcrossing* **by** *force*
hence *downcrossing* (*Suc n*) *ω* ≠ *upcrossing* (*Suc n*) *ω* **unfolding** *stopped-value-def* **by** *force*
thus *?thesis* **using** *upcrossing-le-downcrossing* **by** (*simp add*: *le-neq-implies-less*)
qed

lemma *downcrossing-less-upcrossing*:
assumes *a* < *b upcrossing* (*Suc n*) *ω* ≠ *N*
shows *downcrossing n ω* < *upcrossing* (*Suc n*) *ω*
proof −
have *downcrossing n ω* ≠ *N* **using** *assms* **by** (*metis le-antisym upcrossing-le downcrossing-le-upcrossing-Suc*)
hence *stopped-value X* (*downcrossing n*) *ω* < *stopped-value X* (*upcrossing* (*Suc n*)) *ω*
using *assms stopped-value-downcrossing stopped-value-upcrossing* **by** *force*
hence *downcrossing n ω* ≠ *upcrossing* (*Suc n*) *ω* **unfolding** *stopped-value-def* **by** *force*
thus *?thesis* **using** *downcrossing-le-upcrossing-Suc* **by** (*simp add*: *le-neq-implies-less*)
qed

lemma *upcrossing-less-Suc*:
assumes *a* < *b upcrossing n ω* ≠ *N*
shows *upcrossing n ω* < *upcrossing* (*Suc n*) *ω*
by (*metis assms upcrossing-le-downcrossing downcrossing-less-upcrossing order-le-less-trans le-neq-implies-less upcrossing-le*)

lemma *upcrossing-eq-bound*:
assumes *a* < *b n* ≥ *N*
shows *upcrossing n ω* = *N*

proof −
have ∗: *upcrossing N ω = N*
proof −
{
assume ∗: *upcrossing N ω ≠ N*
hence *asm*: *upcrossing n ω < N* **if** *n ≤ N* **for** *n* **using** *upcrossing-mono upcrossing-le that* **by** (*metis le-antisym le-neq-implies-less*)
{
fix *i j*
assume *i ≤ N i < j*
hence *upcrossing i ω ≠ upcrossing j ω* **by** (*metis Suc-leI asm assms*(*1*) *leD upcrossing-less-Suc upcrossing-mono*)
}
moreover
{
fix *j*
assume *j ≤ N*
hence *upcrossing j ω ≤ upcrossing N ω* **using** *upcrossing-mono* **by** *blast*
hence *upcrossing* (*Suc N*) *ω ≠ upcrossing j ω* **using** *upcrossing-less-Suc*[*OF assms*(*1*) ∗] **by** *simp*
}
ultimately have *inj-on* (*λn. upcrossing n ω*) {*..Suc N*} **unfolding** *inj-on-def* **by** (*metis atMost-iff le-SucE linorder-less-linear*)
hence *card* ((*λn. upcrossing n ω*) ‘ {*..Suc N*}) = *Suc* (*Suc N*) **by** (*simp add*: *inj-on-iff-eq-card*[*THEN iffD1*])
moreover have (*λn. upcrossing n ω*) ‘ {*..Suc N*} ⊆ {*..N*} **using** *upcrossing-le* **by** *blast*
moreover have *card* ((*λn. upcrossing n ω*) ‘ {*..Suc N*}) ≤ *Suc N* **using** *card-mono*[*OF - calculation*(*2*)] **by** *simp*
ultimately have *False* **by** *linarith*
}
thus *?thesis* **by** *blast*
qed
thus *?thesis* **using** *upcrossing-mono*[*OF assms*(*2*), *of ω*] *upcrossing-le*[*of n ω*] **by** *simp*
qed

lemma *downcrossing-eq-bound*:
assumes *a < b n ≥ N*
shows *downcrossing n ω = N*
using *upcrossing-le-downcrossing*[*of n ω*] *downcrossing-le*[*of n ω*] *upcrossing-eq-bound*[*OF assms*] **by** *simp*

lemma *stopping-time-crossings*:
assumes *adapted-process M F 0 X*
shows *stopping-time* (*upcrossing n*) *stopping-time* (*downcrossing n*)
proof −
have *stopping-time* (*upcrossing n*) ∧ *stopping-time* (*downcrossing n*)
proof (*induction n*)

case *0*
then show *?case* **unfolding** *upcrossing-simps downcrossing-simps*
using *stopping-time-const stopping-time-hitting-time*[*OF assms*] **by** *simp*
next
case (*Suc n*)
have *stopping-time* (*upcrossing* (*Suc n*)) **unfolding** *upcrossing-simps*
using *assms Suc downcrossing-le* **by** (*intro stopping-time-hitting-time′*) *auto*
moreover have *stopping-time* (*downcrossing* (*Suc n*)) **unfolding** *downcrossing-simps*
using *assms calculation upcrossing-le* **by** (*intro stopping-time-hitting-time′*) *auto*
ultimately show *?case* **by** *blast*
qed
thus *stopping-time* (*upcrossing n*) *stopping-time* (*downcrossing n*) **by** *blast+*
qed

lemmas *stopping-time-upcrossing* = *stopping-time-crossings*(*1*)
lemmas *stopping-time-downcrossing* = *stopping-time-crossings*(*2*)

— We define *upcrossings-before* as the number of upcrossings which take place strictly before time *N*.

definition *upcrossings-before* :: *′a* ⇒ *nat* **where**
upcrossings-before = (*λω. Sup* {*n. upcrossing n ω* < *N*})

lemma *upcrossings-before-bdd-above*:
assumes *a* < *b*
shows *bdd-above* {*n. upcrossing n ω* < *N*}
proof −
have {*n. upcrossing n ω* < *N*} ⊆ {..<*N*} **unfolding** *lessThan-def Collect-mono-iff*
using *upcrossing-eq-bound*[*OF assms*] *linorder-not-less order-less-irrefl* **by** *metis*
thus *?thesis* **by** (*meson bdd-above-Iio bdd-above-mono*)
qed

lemma *upcrossings-before-less*:
assumes *a* < *b* *0* < *N*
shows *upcrossings-before ω* < *N*
proof −
have ∗: {*n. upcrossing n ω* < *N*} ⊆ {..<*N*} **unfolding** *lessThan-def Collect-mono-iff*
using *upcrossing-eq-bound*[*OF assms*(*1*)] *linorder-not-less order-less-irrefl* **by** *metis*
have *upcrossing 0 ω* < *N* **unfolding** *upcrossing-simps* **by** (*rule assms*)
moreover have *Sup* {..<*N*} < *N* **unfolding** *Sup-nat-def* **using** *assms* **by** *simp*
ultimately show *?thesis* **unfolding** *upcrossings-before-def* **using** *cSup-subset-mono*[*OF - - ∗*] **by** *force*
qed

lemma *upcrossings-before-less-implies-crossing-eq-bound*:

assumes *a* < *b* *upcrossings-before* *ω* < *n*
shows *upcrossing n ω* = *N*
downcrossing n ω = *N*
proof −
have ¬ *upcrossing n ω* < *N* **using** *assms upcrossings-before-bdd-above*[*of ω*] *upcrossings-before-def bdd-above-nat finite-Sup-less-iff* **by** *fastforce*
thus *upcrossing n ω* = *N* **using** *upcrossing-le*[*of n ω*] **by** *simp*
thus *downcrossing n ω* = *N* **using** *upcrossing-le-downcrossing*[*of n ω*] *downcrossing-le*[*of n ω*] **by** *simp*
qed

lemma *upcrossings-before-le*:
assumes *a* < *b*
shows *upcrossings-before ω* ≤ *N*
using *upcrossings-before-less assms less-le-not-le upcrossings-before-def*
by (*cases N*) *auto*

lemma *upcrossings-before-mem*:
assumes *a* < *b* *0* < *N*
shows *upcrossings-before ω* ∈ {*n*. *upcrossing n ω* < *N*} ∩ {..<*N*}
proof −
have *upcrossing 0 ω* < *N* **using** *assms* **unfolding** *upcrossing-simps* **by** *simp*
hence {*n*. *upcrossing n ω* < *N*} ≠ {} **by** *blast*
moreover have *finite* {*n*. *upcrossing n ω* < *N*} **using** *upcrossings-before-bdd-above*[*OF assms*(*1*)] **by** (*simp add*: *bdd-above-nat*)
ultimately show *?thesis* **using** *Max-in upcrossings-before-less*[*OF assms*(*1*,*2*)] *Sup-nat-def upcrossings-before-def* **by** *auto*
qed

lemma *upcrossing-less-of-le-upcrossings-before*:
assumes *a* < *b* *0* < *N* *n* ≤ *upcrossings-before ω*
shows *upcrossing n ω* < *N*
using *upcrossings-before-mem*[*OF assms*(*1*,*2*), *of ω*] *upcrossing-mono*[*OF assms*(*3*), *of ω*] **by** *simp*

lemma *upcrossings-before-sum-def*:
assumes *a* < *b*
shows *upcrossings-before ω* = (∑ *k*∈{*1*..*N*}. *indicator* {*n*. *upcrossing n ω* < *N*} *k*)
proof (*cases N*)
case *0*
then show *?thesis* **unfolding** *upcrossings-before-def* **by** *simp*
next
case (*Suc N′*)
have *upcrossing 0 ω* < *N* **using** *assms Suc* **unfolding** *upcrossing-simps* **by** *simp*
hence {*n*. *upcrossing n ω* < *N*} ≠ {} **by** *blast*
hence ∗: ¬ *upcrossing n ω* < *N* **if** *n* ∈ {*upcrossings-before ω* <..*N*} **for** *n*
using *finite-Sup-less-iff*[*THEN iffD1*, *OF bdd-above-nat*[*THEN iffD1*, *OF upcrossings-before-bdd-above*], *of ω n*]

by (*metis that assms greaterThanAtMost-iff less-not-refl mem-Collect-eq upcrossings-before-def*)
have $**$: *upcrossing n* ω < *N* **if** *n* ∈ {*1..upcrossings-before* ω} **for** *n*
using *assms that Suc* **by** (*intro upcrossing-less-of-le-upcrossings-before*) *auto*
have *upcrossings-before* ω < *N* **using** *upcrossings-before-less Suc assms* **by** *simp*
hence {*1..N*} − {*1..upcrossings-before* ω} = {*upcrossings-before* ω<..*N*}
{*1..N*} ∩ {*1..upcrossings-before* ω} = {*1..upcrossings-before* ω} **by** *force+*
hence ($\sum$ *k*∈{*1..N*}. *indicator* {*n. upcrossing n* ω < *N*} *k*) =
($\sum$ *k*∈{*1..upcrossings-before* ω}. *indicator* {*n. upcrossing n* ω < *N*} *k*) + ($\sum$ *k*∈{*upcrossings-before* ω <..*N*}. *indicator* {*n. upcrossing n* ω < *N*} *k*)
using *sum.Int-Diff*[*OF finite-atLeastAtMost*, *of - 1 N* {*1..upcrossings-before* ω}] **by** *metis*
also have ... = *upcrossings-before* ω **using** $*$ $**$ **by** *simp*
finally show *?thesis* **by** *argo*
qed

lemma *upcrossings-before-measurable*:
assumes *adapted-process M F 0 X a* < *b*
shows *upcrossings-before* ∈ *borel-measurable M*
unfolding *upcrossings-before-sum-def*[*OF assms*(*2*)]
using *stopping-time-measurable*[*OF stopping-time-crossings*(*1*), *OF assms*(*1*)] **by** *simp*

lemma *upcrossings-before-measurable'*:
assumes *adapted-process M F 0 X a* < *b*
shows ($\lambda\omega$. *real* (*upcrossings-before* ω)) ∈ *borel-measurable M*
using *real-embedding-borel-measurable upcrossings-before-measurable*[*OF assms*] **by** *simp*

end

lemma *crossing-eq-crossing*:
assumes *N* ≤ *N'*
and *downcrossing X a b N n* ω < *N*
shows *upcrossing X a b N n* ω = *upcrossing X a b N' n* ω
downcrossing X a b N n ω = *downcrossing X a b N' n* ω
proof −
have *upcrossing X a b N n* ω = *upcrossing X a b N' n* ω ∧ *downcrossing X a b N n* ω = *downcrossing X a b N' n* ω **using** *assms*(*2*)
proof (*induction n*)
case *0*
show *?case* **by** (*metis* (*no-types*, *lifting*) *upcrossing-simps*(*1*) *assms atLeast-0 bot-nat-0.extremum hitting-time-def hitting-time-eq-hitting-time inf-top.right-neutral leD downcrossing-mono downcrossing-simps*(*1*) *max-nat.left-neutral*)
next
case (*Suc n*)
hence *upper-less*: *upcrossing X a b N* (*Suc n*) ω < *N* **using** *upcrossing-le-downcrossing Suc order.strict-trans1* **by** *blast*
hence *lower-less*: *downcrossing X a b N n* ω < *N* **using** *downcrossing-le-upcrossing-Suc*

order.strict-trans1 **by** *blast*

obtain *j* **where** *j* ∈ {*downcrossing X a b N n* ω..<*N*} *X j* ω ∈ {*b*..}
using *hitting-time-less-iff*[*THEN iffD1*, *OF order-refl*] *upper-less* **by** (*force simp add*: *upcrossing-simps*)
hence *upper-eq*: *upcrossing X a b N* (*Suc n*) ω = *upcrossing X a b N′* (*Suc n*) ω
using *Suc*(*1*)[*OF lower-less*] *assms*(*1*)
by (*auto simp add*: *upcrossing-simps intro*!: *hitting-time-eq-hitting-time*)
obtain *j* **where** *j*: *j* ∈ {*upcrossing X a b N* (*Suc n*) ω..<*N*} *X j* ω ∈ {..*a*}
using *Suc*(*2*) *hitting-time-less-iff*[*THEN iffD1*, *OF order-refl*] **by** (*force simp add*: *downcrossing-simps*)
thus *?case* **unfolding** *downcrossing-simps upper-eq* **by** (*force intro*: *hitting-time-eq-hitting-time assms*)
qed
thus *upcrossing X a b N n* ω = *upcrossing X a b N′ n* ω *downcrossing X a b N n* ω = *downcrossing X a b N′ n* ω **by** *auto*
qed

lemma *crossing-eq-crossing′*:
assumes *N* ≤ *N′*
and *upcrossing X a b N* (*Suc n*) ω < *N*
shows *upcrossing X a b N* (*Suc n*) ω = *upcrossing X a b N′* (*Suc n*) ω
downcrossing X a b N n ω = *downcrossing X a b N′ n* ω
proof −
show *lower-eq*: *downcrossing X a b N n* ω = *downcrossing X a b N′ n* ω
using *downcrossing-le-upcrossing-Suc*[*THEN order.strict-trans1*] *crossing-eq-crossing assms* **by** *fast*
have ∃*j*∈{*downcrossing X a b N n* ω..<*N*}. *X j* ω ∈ {*b*..} **using** *assms*(*2*) **by** (*intro hitting-time-less-iff*[*OF order-refl*, *THEN iffD1*]) (*simp add*: *upcrossing-simps lower-eq*)
then obtain *j* **where** *j*∈{*downcrossing X a b N n* ω..*N*} *X j* ω ∈ {*b*..} **by** *fastforce*
thus *upcrossing X a b N* (*Suc n*) ω = *upcrossing X a b N′* (*Suc n*) ω
unfolding *upcrossing-simps stopped-value-def* **using** *hitting-time-eq-hitting-time*[*OF assms*(*1*)] *lower-eq* **by** *metis*
qed

lemma *upcrossing-eq-upcrossing*:
assumes *N* ≤ *N′*
and *upcrossing X a b N n* ω < *N*
shows *upcrossing X a b N n* ω = *upcrossing X a b N′ n* ω
using *crossing-eq-crossing′*[*OF assms*(*1*)] *assms*(*2*) *upcrossing-simps*
by (*cases n*) (*presburger*, *fast*)

lemma *upcrossings-before-zero*: *upcrossings-before X a b 0* ω = *0*
unfolding *upcrossings-before-def* **by** *simp*

lemma *upcrossings-before-less-exists-upcrossing*:
assumes *a* < *b*

and *upcrossing*: $N \leq L$ X L $\omega < a$ $L \leq U$ $b < X$ U ω
shows *upcrossings-before* X a b N ω < *upcrossings-before* X a b (*Suc* U) ω
proof −
have *upcrossing* X a b (*Suc* U) (*upcrossings-before* X a b N ω) $\omega \leq L$
using *assms upcrossing-le*[*THEN order-trans*, *OF upcrossing*(*1*)]
by (*cases* $0 < N$, *subst upcrossing-eq-upcrossing*[*of N Suc U*, *symmetric*, *OF - upcrossing-less-of-le-upcrossings-before*])
(*auto simp add*: *upcrossings-before-zero upcrossing-simps*)
hence *downcrossing* X a b (*Suc* U) (*upcrossings-before* X a b N ω) $\omega \leq U$
unfolding *downcrossing-simps* **using** *upcrossing* **by** (*force intro*: *hitting-time-le-iff*[*THEN iffD2*])
hence *upcrossing* X a b (*Suc* U) (*Suc* (*upcrossings-before* X a b N ω)) ω < *Suc* U
unfolding *upcrossing-simps* **using** *upcrossing* **by** (*force intro*: *hitting-time-less-iff*[*THEN iffD2*])
thus *?thesis* **using** *cSup-upper*[*OF - upcrossings-before-bdd-above*[*OF assms*(*1*)]] *upcrossings-before-def* **by** *fastforce*
qed

lemma *crossings-translate*:
upcrossing X a b N = *upcrossing* (λn ω. (X n ω + c)) (a + c) (b + c) N
downcrossing X a b N = *downcrossing* (λn ω. (X n ω + c)) (a + c) (b + c) N
proof −
have *upper*: *upcrossing* X a b N n = *upcrossing* (λn ω. (X n ω + c)) (a + c) (b + c) N n **for** n
proof (*induction* n)
case *0*
then show *?case* **by** (*simp only*: *upcrossing.simps*)
next
case (*Suc* n)
have ((+) c ‘ {..a}) = {..a + c} **by** *simp*
moreover have ((+) c ‘ {b..}) = {b + c..} **by** *simp*
ultimately show *?case* **unfolding** *upcrossing.simps* **using** *hitting-time-translate*[*of* X {b..} c] *hitting-time-translate*[*of* X {..a} c] *Suc* **by** *presburger*
qed
thus *upcrossing* X a b N = *upcrossing* (λn ω. (X n ω + c)) (a + c) (b + c) N **by** *blast*
have ((+) c ‘ {..a}) = {..a + c} **by** *simp*
thus *downcrossing* X a b N = *downcrossing* (λn ω. (X n ω + c)) (a + c) (b + c) N **using** *upper downcrossing-simps hitting-time-translate*[*of* X {..a} c] **by** *presburger*
qed

lemma *upcrossings-before-translate*:
upcrossings-before X a b N = *upcrossings-before* (λn ω. (X n ω + c)) (a + c) (b + c) N
using *upcrossings-before-def crossings-translate* **by** *simp*

lemma *crossings-pos-eq*:

assumes *a* < *b*
shows *upcrossing X a b N* = *upcrossing* (λ*n* ω. *max 0* (*X n* ω − *a*)) *0* (*b* − *a*) *N*
downcrossing X a b N = *downcrossing* (λ*n* ω. *max 0* (*X n* ω − *a*)) *0* (*b* − *a*) *N*
proof −
have ∗: *max 0* (*x* − *a*) ∈ {..*0*} ⟷ *x* − *a* ∈ {..*0*} *max 0* (*x* − *a*) ∈ {*b* − *a*..} ⟷ *x* − *a* ∈ {*b* − *a*..} **for** *x* **using** *assms* **by** *auto*
have *upcrossing X a b N* = *upcrossing* (λ*n* ω. *X n* ω − *a*) *0* (*b* − *a*) *N* **using** *crossings-translate*[*of X a b N* − *a*] **by** *simp*
thus *upper*: *upcrossing X a b N* = *upcrossing* (λ*n* ω. *max 0* (*X n* ω − *a*)) *0* (*b* − *a*) *N* **unfolding** *upcrossing-def hitting-time-def*′ **using** ∗ **by** *presburger*

thus *downcrossing X a b N* = *downcrossing* (λ*n* ω. *max 0* (*X n* ω − *a*)) *0* (*b* − *a*) *N*
unfolding *downcrossing-simps hitting-time-def*′ **using** *upper* ∗ **by** *simp*
qed

lemma *upcrossings-before-mono*:
assumes *a* < *b N* ≤ *N*′
shows *upcrossings-before X a b N* ω ≤ *upcrossings-before X a b N*′ ω
proof (*cases N*)
case *0*
then show *?thesis* **unfolding** *upcrossings-before-def* **by** *simp*
next
case (*Suc N*′)
hence *upcrossing X a b N 0* ω < *N* **unfolding** *upcrossing-simps* **by** *simp*
thus *?thesis* **unfolding** *upcrossings-before-def* **using** *upcrossings-before-bdd-above upcrossing-eq-upcrossing assms* **by** (*intro cSup-subset-mono*) *auto*
qed

lemma *upcrossings-before-pos-eq*:
assumes *a* < *b*
shows *upcrossings-before X a b N* = *upcrossings-before* (λ*n* ω. *max 0* (*X n* ω − *a*)) *0* (*b* − *a*) *N*
using *upcrossings-before-def crossings-pos-eq*[*OF assms*] **by** *simp*

— We define *upcrossings* to be the total number of upcrossings a stochastic process completes as $N \longrightarrow \infty$.

definition *upcrossings* :: (*nat* ⇒ ′*a* ⇒ *real*) ⇒ *real* ⇒ *real* ⇒ ′*a* ⇒ *ennreal* **where**
upcrossings X a b = (λω. (*SUP N*. *ennreal* (*upcrossings-before X a b N* ω)))

lemma *upcrossings-measurable*:
assumes *adapted-process M F 0 X a* < *b*
shows *upcrossings X a b* ∈ *borel-measurable M*
unfolding *upcrossings-def*
using *upcrossings-before-measurable*′[*OF assms*] **by** (*auto intro*!: *borel-measurable-SUP*)

end

lemma (**in** *nat-finite-filtered-measure*) *integrable-upcrossings-before*:
assumes *adapted-process M F 0 X a* $<$ *b*
shows *integrable M* ($\lambda\omega$. *real* (*upcrossings-before X a b N* ω))
proof $-$
have ($\int^+$ *x*. *ennreal* (*norm* (*real* (*upcrossings-before X a b N x*))) ∂M) $\leq$ ($\int^+$ *x*. *ennreal N* ∂M) **using** *upcrossings-before-le*[*OF assms*(*2*)] **by** (*intro nn-integral-mono*) *simp*
also have ... = *ennreal N* $*$ *emeasure M* (*space M*) **by** *simp*
also have ... $< \infty$ **by** (*metis emeasure-real ennreal-less-top ennreal-mult-less-top infinity-ennreal-def*)
finally show *?thesis* **by** (*intro integrableI-bounded upcrossings-before-measurable′ assms*)
qed

3.2 Doob's Upcrossing Inequality

Doob's upcrossing inequality provides a bound on the expected number of upcrossings a submartingale completes before some point in time. The proof follows the proof presented in the paper *A Formalization of Doob′s Martingale Convergence Theorems in mathlib* [1] [2].

context *nat-finite-filtered-measure*
begin

theorem *upcrossing-inequality*:
fixes *a b* :: *real* **and** *N* :: *nat*
assumes *submartingale M F 0 X*
shows ($b - a$) $*$ ($\int\omega$. *real* (*upcrossings-before X a b N* ω) ∂M) $\leq$ ($\int\omega$. *max 0* (*X N* $\omega - a$) ∂M)
proof $-$
interpret *submartingale-linorder M F 0 X* **unfolding** *submartingale-linorder-def* **by** (*intro assms*)
show *?thesis*
proof (*cases a* $<$ *b*)
case *True*
— We show the statement first for *X 0* non-negative and *X N* greater than or equal to *a*.
have $*$: ($b - a$) $*$ ($\int\omega$. *real* (*upcrossings-before X a b N* ω) ∂M) $\leq$ ($\int\omega$. *X N* ω ∂M)
if *asm*: *submartingale M F 0 X a* $<$ *b* $\bigwedge\omega$. *X 0* $\omega \geq 0$ $\bigwedge\omega$. *X N* $\omega \geq a$
for *a b X*
proof $-$
interpret *subm*: *submartingale M F 0 X* **by** (*intro asm*)
define *C* :: *nat* $\Rightarrow$ *′a* $\Rightarrow$ *real* **where** *C* = ($\lambda n\ \omega$. $\sum k < N$. *indicator* {*downcrossing X a b N k* ω..<*upcrossing X a b N* (*Suc k*) ω} *n*)
have *C-values*: *C n* $\omega \in$ {*0*, *1*} **for** *n* ω
proof (*cases* $\exists j < N$. $n \in$ {*downcrossing X a b N j* ω..<*upcrossing X a b N* (*Suc j*) ω})

case *True*
then obtain *j* **where** *j*: $j \in \{..<N\}$ $n \in \{downcrossing\ X\ a\ b\ N\ j\ \omega..<upcrossing\ X\ a\ b\ N\ (Suc\ j)\ \omega\}$ **by** *blast*
{
fix *k l* :: *nat* **assume** *k-less-l*: $k < l$
hence *Suc-k-le-l*: $Suc\ k \leq l$ **by** *simp*
have $\{downcrossing\ X\ a\ b\ N\ k\ \omega..<upcrossing\ X\ a\ b\ N\ (Suc\ k)\ \omega\} \cap \{downcrossing\ X\ a\ b\ N\ l\ \omega..<upcrossing\ X\ a\ b\ N\ (Suc\ l)\ \omega\} = \{downcrossing\ X\ a\ b\ N\ l\ \omega..<upcrossing\ X\ a\ b\ N\ (Suc\ k)\ \omega\}$
using *k-less-l upcrossing-mono downcrossing-mono* **by** *simp*
moreover have $upcrossing\ X\ a\ b\ N\ (Suc\ k)\ \omega \leq downcrossing\ X\ a\ b\ N\ l\ \omega$
using *upcrossing-le-downcrossing downcrossing-mono*[*OF Suc-k-le-l*] *order-trans* **by** *blast*
ultimately have $\{downcrossing\ X\ a\ b\ N\ k\ \omega..<upcrossing\ X\ a\ b\ N\ (Suc\ k)\ \omega\} \cap \{downcrossing\ X\ a\ b\ N\ l\ \omega..<upcrossing\ X\ a\ b\ N\ (Suc\ l)\ \omega\} = \{\}$ **by** *simp*
}
hence *disjoint-family-on* $(\lambda k.\ \{downcrossing\ X\ a\ b\ N\ k\ \omega..<upcrossing\ X\ a\ b\ N\ (Suc\ k)\ \omega\})\ \{..<N\}$
unfolding *disjoint-family-on-def*
by (*metis Int-commute linorder-less-linear*)
hence $C\ n\ \omega = 1$ **unfolding** *C-def* **using** *sum-indicator-disjoint-family*[**where** $?f=\lambda\text{-}.\ 1$] *j* **by** *fastforce*
thus *?thesis* **by** *blast*
next
case *False*
hence $C\ n\ \omega = 0$ **unfolding** *C-def* **by** *simp*
thus *?thesis* **by** *simp*
qed
hence *C-interval*: $C\ n\ \omega \in \{0..1\}$ **for** *n ω* **by** (*metis atLeastAtMost-iff empty-iff insert-iff order.refl zero-less-one-class.zero-le-one*)

— We consider the discrete stochastic integral of C and $\lambda n\ \omega.\ 1 - C\ n\ \omega$.
define C' **where** $C' = (\lambda n\ \omega.\ \sum k < n.\ C\ k\ \omega *_R (X\ (Suc\ k)\ \omega - X\ k\ \omega))$
define *one-minus-C'* **where** $one\text{-}minus\text{-}C' = (\lambda n\ \omega.\ \sum k < n.\ (1 - C\ k\ \omega) *_R (X\ (Suc\ k)\ \omega - X\ k\ \omega))$

— We use the fact that the crossing times are stopping times to show that C is predictable.
have *adapted-C*: *adapted-process M F 0 C*
proof
fix *i*
have $(\lambda\omega.\ indicat\text{-}real\ \{downcrossing\ X\ a\ b\ N\ k\ \omega..<upcrossing\ X\ a\ b\ N\ (Suc\ k)\ \omega\}\ i) \in borel\text{-}measurable\ (F\ i)$ **for** *k*
unfolding *indicator-def*
using *stopping-time-upcrossing*[*OF subm.adapted-process-axioms*, *THEN stopping-time-measurable-gr*]
stopping-time-downcrossing[*OF subm.adapted-process-axioms*, *THEN stopping-time-measurable-le*]
by *force*

thus *C i* ∈ *borel-measurable* (*F i*) **unfolding** *C-def* **by** *simp*
qed
hence *adapted-process M F 0* (λ*n* ω. *1* − *C n* ω) **by** (*intro adapted-process.diff-adapted adapted-process-const*)
hence *submartingale-one-minus-C′*: *submartingale M F 0 one-minus-C′* **unfolding** *one-minus-C′-def* **using** *C-interval*
by (*intro submartingale-partial-sum-scaleR*[*of - - 1*] *submartingale-linorder.intro asm*) *auto*

have *C n* ∈ *borel-measurable M* **for** *n*
using *adapted-C adapted-process.adapted measurable-from-subalg subalg* **by** *blast*

have *integrable-C′*: *integrable M* (*C′ n*) **for** *n* **unfolding** *C′-def* **using** *C-interval*
by (*intro submartingale-partial-sum-scaleR*[*THEN submartingale.integrable*] *submartingale-linorder.intro adapted-C asm*) *auto*

— We show the following inequality, by using the fact that *one-minus-C′* is a submartingale.
have $integral^L$ *M* (*C′ n*) ≤ $integral^L$ *M* (*X n*) **for** *n*
proof −
interpret *subm′*: *submartingale-linorder M F 0 one-minus-C′* **unfolding** *submartingale-linorder-def* **by** (*rule submartingale-one-minus-C′*)
have *0* ≤ $integral^L$ *M* (*one-minus-C′ n*)
using *subm′.set-integral-le*[*OF sets.top*, **where** *i=0* **and** *j=n*] *space-F subm′.integrable* **by** (*fastforce simp add*: *set-integral-space one-minus-C′-def*)
moreover have *one-minus-C′ n* ω = (∑*k* < *n*. *X* (*Suc k*) ω − *X k* ω) − *C′ n* ω **for** ω
unfolding *one-minus-C′-def C′-def* **by** (*simp only*: *scaleR-diff-left sum-subtractf scale-one*)
ultimately have *0* ≤ (*LINT* ω|*M*. (∑*k* < *n*. *X* (*Suc k*) ω − *X k* ω)) − $integral^L$ *M* (*C′ n*)
using *subm.integrable integrable-C′*
by (*subst Bochner-Integration.integral-diff*[*symmetric*]) (*auto simp add*: *one-minus-C′-def*)
moreover have (*LINT* ω|*M*. (∑*k*<*n*. *X* (*Suc k*) ω − *X k* ω)) ≤ (*LINT* ω|*M*. *X n* ω) **using** *asm sum-lessThan-telescope*[*of* λ*i*. *X i* - *n*] *subm.integrable*
by (*intro integral-mono*) *auto*
ultimately show *?thesis* **by** *linarith*
qed
moreover have (*b* − *a*) ∗ (∫ω. *real* (*upcrossings-before X a b N* ω) ∂*M*) ≤ $integral^L$ *M* (*C′ N*)
proof (*cases N*)
case *0*
then show *?thesis* **using** *C′-def upcrossings-before-zero* **by** *simp*
next
case (*Suc N′*)
{

fix ω

have *dc-not-N*: *downcrossing* $X\ a\ b\ N\ k\ \omega \neq N$ **if** $k <$ *upcrossings-before* $X\ a\ b\ N\ \omega$ **for** k

by (*metis Suc Suc-leI asm*(*2*) *downcrossing-le-upcrossing-Suc leD that upcrossing-less-of-le-upcrossings-before zero-less-Suc*)

have *uc-not-N*:*upcrossing* $X\ a\ b\ N\ (Suc\ k)\ \omega \neq N$ **if** $k <$ *upcrossings-before* $X\ a\ b\ N\ \omega$ **for** k

by (*metis Suc Suc-leI asm*(*2*) *order-less-irrefl that upcrossing-less-of-le-upcrossings-before zero-less-Suc*)

have *subset-lessThan-N*: $\{$*downcrossing* $X\ a\ b\ N\ i\ \omega..<$*upcrossing* $X\ a\ b\ N\ (Suc\ i)\ \omega\} \subseteq \{..<N\}$ **if** $i < N$ **for** i **using** *that*

by (*simp add*: *lessThan-atLeast0 upcrossing-le*)

— First we rewrite the sum as follows:

have $C'\ N\ \omega = (\sum k<N.\ \sum i<N.$ *indicator* $\{$*downcrossing* $X\ a\ b\ N\ i\ \omega..<$*upcrossing* $X\ a\ b\ N\ (Suc\ i)\ \omega\}\ k * (X\ (Suc\ k)\ \omega - X\ k\ \omega))$

unfolding *C′-def C-def* **by** (*simp add*: *sum-distrib-right*)

also have ... $= (\sum i<N.\ \sum k<N.$ *indicator* $\{$*downcrossing* $X\ a\ b\ N\ i\ \omega..<$*upcrossing* $X\ a\ b\ N\ (Suc\ i)\ \omega\}\ k * (X\ (Suc\ k)\ \omega - X\ k\ \omega))$

using *sum.swap* **by** *fast*

also have ... $= (\sum i<N.\ \sum k\in\{..<N\} \cap \{$*downcrossing* $X\ a\ b\ N\ i\ \omega..<$*upcrossing* $X\ a\ b\ N\ (Suc\ i)\ \omega\}.\ X\ (Suc\ k)\ \omega - X\ k\ \omega)$

by (*subst Indicator-Function.sum-indicator-mult*) *simp+*

also have ... $= (\sum i<N.\ \sum k\in\{$*downcrossing* $X\ a\ b\ N\ i\ \omega..<$*upcrossing* $X\ a\ b\ N\ (Suc\ i)\ \omega\}.\ X\ (Suc\ k)\ \omega - X\ k\ \omega)$

using *subset-lessThan-N*[*THEN Int-absorb1*] **by** *simp*

also have ... $= (\sum i<N.\ X\ ($*upcrossing* $X\ a\ b\ N\ (Suc\ i)\ \omega)\ \omega - X\ ($*downcrossing* $X\ a\ b\ N\ i\ \omega)\ \omega)$

by (*subst sum-Suc-diff′*[*OF downcrossing-le-upcrossing-Suc*]) *blast*

finally have $*$: $C'\ N\ \omega = (\sum i<N.\ X\ ($*upcrossing* $X\ a\ b\ N\ (Suc\ i)\ \omega)\ \omega - X\ ($*downcrossing* $X\ a\ b\ N\ i\ \omega)\ \omega)$ **.**

— For $k \leq N$, we consider three cases:

— 1. If $k <$ *upcrossings-before* $X\ a\ b\ N\ \omega$, then $X\ ($*upcrossing* $X\ a\ b\ N\ (Suc\ k)\ \omega)\ \omega - X\ ($*downcrossing* $X\ a\ b\ N\ k\ \omega)\ \omega \geq b - a$

— 2. If *upcrossings-before* $X\ a\ b\ N\ \omega < k$, then $X\ ($*upcrossing* $X\ a\ b\ N\ (Suc\ k)\ \omega)\ \omega = X\ ($*downcrossing* $X\ a\ b\ N\ k\ \omega)\ \omega$

— 3. If $k =$ *upcrossings-before* $X\ a\ b\ N\ \omega$, then $X\ ($*upcrossing* $X\ a\ b\ N\ (Suc\ k)\ \omega)\ \omega - X\ ($*downcrossing* $X\ a\ b\ N\ k\ \omega)\ \omega \geq 0$

have *summand-zero-if*: $X\ ($*upcrossing* $X\ a\ b\ N\ (Suc\ k)\ \omega)\ \omega - X\ ($*downcrossing* $X\ a\ b\ N\ k\ \omega)\ \omega = 0$ **if** $k >$ *upcrossings-before* $X\ a\ b\ N\ \omega$ **for** k

using *that upcrossings-before-less-implies-crossing-eq-bound*[*OF asm*(*2*)] **by** *simp*

have *summand-nonneg-if*: $X\ ($*upcrossing* $X\ a\ b\ N\ (Suc\ ($*upcrossings-before* $X\ a\ b\ N\ \omega))\ \omega)\ \omega - X\ ($*downcrossing* $X\ a\ b\ N\ ($*upcrossings-before* $X\ a\ b\ N\ \omega)\ \omega)$

ω ≥ 0
using *upcrossings-before-less-implies-crossing-eq-bound*(*1*)[*OF asm*(*2*) *lessI*]
stopped-value-downcrossing[*of X a b N - ω*, *THEN order-trans*, *OF - asm*(*4*)[*of ω*]]
by (*cases downcrossing X a b N* (*upcrossings-before X a b N ω*) *ω ≠ N*) (*simp add*: *stopped-value-def*)+

have *interval*: {*upcrossings-before X a b N ω*..<*N*} − {*upcrossings-before X a b N ω*} = {*upcrossings-before X a b N ω*<..<*N*}
using *Diff-insert atLeastSucLessThan-greaterThanLessThan lessThan-Suc lessThan-minus-lessThan* **by** *metis*

have (*b* − *a*) ∗ *real* (*upcrossings-before X a b N ω*) = (∑ -<*upcrossings-before X a b N ω*. *b* − *a*) **by** *simp*
also have ... ≤ (∑ *k*<*upcrossings-before X a b N ω*. *stopped-value X* (*upcrossing X a b N* (*Suc k*)) *ω* − *stopped-value X* (*downcrossing X a b N k*) *ω*)
using *stopped-value-downcrossing*[*OF dc-not-N*] *stopped-value-upcrossing*[*OF uc-not-N*] **by** (*force intro*!: *sum-mono*)
also have ... = (∑ *k*<*upcrossings-before X a b N ω*. *X* (*upcrossing X a b N* (*Suc k*) *ω*) *ω* − *X* (*downcrossing X a b N k ω*) *ω*) **unfolding** *stopped-value-def* **by** *blast*
also have ... ≤ (∑ *k*<*upcrossings-before X a b N ω*. *X* (*upcrossing X a b N* (*Suc k*) *ω*) *ω* − *X* (*downcrossing X a b N k ω*) *ω*)
\+ (∑ *k*∈{*upcrossings-before X a b N ω*}. *X* (*upcrossing X a b N* (*Suc k*) *ω*) *ω* − *X* (*downcrossing X a b N k ω*) *ω*)
\+ (∑ *k*∈{*upcrossings-before X a b N ω*<..<*N*}. *X* (*upcrossing X a b N* (*Suc k*) *ω*) *ω* − *X* (*downcrossing X a b N k ω*) *ω*)
using *summand-zero-if summand-nonneg-if* **by** *auto*
also have ... = (∑ *k*<*N*. *X* (*upcrossing X a b N* (*Suc k*) *ω*) *ω* − *X* (*downcrossing X a b N k ω*) *ω*)
using *upcrossings-before-le*[*OF asm*(*2*)]
by (*subst sum.subset-diff*[**where** *A*={..<*N*} **and** *B*={..<*upcrossings-before X a b N ω*}], *simp*, *simp*,
subst sum.subset-diff[**where** *A*={..<*N*} − {..<*upcrossings-before X a b N ω*} **and** *B*={*upcrossings-before X a b N ω*}])
(*simp add*: *Suc asm*(*2*) *upcrossings-before-less*, *simp*, *simp add*: *interval*)
finally have (*b* − *a*) ∗ *real* (*upcrossings-before X a b N ω*) ≤ *C′ N ω* **using** ∗ **by** *presburger*
}
thus *?thesis* **using** *integrable-upcrossings-before subm.adapted-process-axioms asm integrable-C′*
by (*subst integral-mult-right-zero*[*symmetric*], *intro integral-mono*) *auto*
qed
ultimately show *?thesis* **using** *order-trans* **by** *blast*
qed

have (*b* − *a*) ∗ (∫*ω*. *real* (*upcrossings-before X a b N ω*) ∂*M*) = (*b* − *a*) ∗ (∫*ω*. *real* (*upcrossings-before* (*λn ω*. *max 0* (*X n ω* − *a*)) *0* (*b* − *a*) *N ω*) ∂*M*)

using *upcrossings-before-pos-eq*[*OF True*] **by** *simp*
also have ... $\leq$ ($\int\omega$. *max 0* (*X N* ω − *a*) ∂M)
using $*$[*OF submartingale-linorder.max-0*[*OF submartingale-linorder.intro*, *OF submartingale.diff*, *OF assms supermartingale-const*], *of 0 b* − *a a*] *True* **by** *simp*
finally show *?thesis* **.**
next
case *False*
have *0* $\leq$ ($\int\omega$. *max 0* (*X N* ω − *a*) ∂M) **by** *simp*
moreover have *0* $\leq$ ($\int\omega$. *real* (*upcrossings-before X a b N* ω) ∂M) **by** *simp*
moreover have *b* − *a* $\leq$ *0* **using** *False* **by** *simp*
ultimately show *?thesis* **using** *mult-nonpos-nonneg order-trans* **by** *meson*
qed
qed

theorem *upcrossing-inequality-Sup*:
fixes *a b* :: *real*
assumes *submartingale M F 0 X*
shows (*b* − *a*) $*$ ($\int^+\omega$. *upcrossings X a b* ω ∂M) $\leq$ (*SUP N*. ($\int^+\omega$. *max 0* (*X N* ω − *a*) ∂M))
proof −
interpret *submartingale M F 0 X* **by** (*intro assms*)
show *?thesis*
proof (*cases a* < *b*)
case *True*
have ($\int^+\omega$. *upcrossings X a b* ω ∂M) = (*SUP N*. ($\int^+\omega$. *real* (*upcrossings-before X a b N* ω) ∂M))
unfolding *upcrossings-def*
using *upcrossings-before-mono True upcrossings-before-measurable'*[*OF adapted-process-axioms*]
by (*auto intro*: *nn-integral-monotone-convergence-SUP simp add*: *mono-def le-funI*)
hence (*b* − *a*) $*$ ($\int^+\omega$. *upcrossings X a b* ω ∂M) = (*SUP N*. (*b* − *a*) $*$ ($\int^+\omega$. *real* (*upcrossings-before X a b N* ω) ∂M))
by (*simp add*: *SUP-mult-left-ennreal*)
moreover
{
fix *N*
have ($\int^+\omega$. *real* (*upcrossings-before X a b N* ω) ∂M) = ($\int\omega$. *real* (*upcrossings-before X a b N* ω) ∂M)
by (*force intro*!: *nn-integral-eq-integral integrable-upcrossings-before True adapted-process-axioms*)
moreover have ($\int^+\omega$. *max 0* (*X N* ω − *a*) ∂M) = ($\int\omega$. *max 0* (*X N* ω − *a*) ∂M)
using *Bochner-Integration.integrable-diff*[*OF integrable integrable-const*]
by (*force intro*!: *nn-integral-eq-integral*)
ultimately have (*b* − *a*) $*$ ($\int^+\omega$. *real* (*upcrossings-before X a b N* ω) ∂M) $\leq$ ($\int^+\omega$. *max 0* (*X N* ω − *a*) ∂M)
using *upcrossing-inequality*[*OF assms*, *of b a N*] *True ennreal-mult'*[*symmetric*] **by** *simp*

}
ultimately show *?thesis* **by** (*force intro*!: *Sup-mono*)
qed (*simp add*: *ennreal-neg*)
qed

end

end

4 Doob's First Martingale Convergence Theorem

theory *Doob-Convergence*
imports *Upcrossing*
begin

context *nat-finite-filtered-measure*
begin

Doob's martingale convergence theorem states that, if we have a submartingale where the supremum over the mean of the positive parts is finite, then the limit process exists almost surely and is integrable. Furthermore, the limit process is measurable with respect to the smallest σ-algebra containing all of the σ-algebras in the filtration. The argumentation below is taken mostly from [3].

theorem *submartingale-convergence-AE*:
fixes $X :: nat \Rightarrow 'a \Rightarrow real$
assumes *submartingale M F 0 X*
and $\bigwedge n.\ (\int \omega.\ max\ 0\ (X\ n\ \omega)\ \partial M) \leq C$
obtains X_{lim} **where** $AE\ \omega\ in\ M.\ (\lambda n.\ X\ n\ \omega) \longrightarrow X_{lim}\ \omega$
$integrable\ M\ X_{lim}$
$X_{lim} \in borel\text{-}measurable\ (F_\infty)$
proof −
interpret *submartingale-linorder M F 0 X* **unfolding** *submartingale-linorder-def*
by (*rule assms*)

— We first show that the number of upcrossings has to be finite using the upcrossing inequality we proved above.

have *finite-upcrossings*: $AE\ \omega\ in\ M.\ upcrossings\ X\ a\ b\ \omega \neq \infty$ **if** $a < b$ **for** $a\ b$
proof −
have *C-nonneg*: $C \geq 0$ **using** *assms(2)* **by** (*meson Bochner-Integration.integral-nonneg linorder-not-less max.cobounded1 order-less-le-trans*)
{
fix n
have $(\int^+\omega.\ max\ 0\ (X\ n\ \omega - a)\ \partial M) \leq (\int^+\omega.\ max\ 0\ (X\ n\ \omega) + |a|\ \partial M)$
by (*fastforce intro*: *nn-integral-mono ennreal-leI*)
also have ... $= (\int^+\omega.\ max\ 0\ (X\ n\ \omega)\ \partial M) + |a| * emeasure\ M\ (space\ M)$
by (*simp add*: *nn-integral-add*)

also have ... = ($\int \omega$. *max 0* (*X n* ω) ∂M) + |*a*| ∗ *emeasure M* (*space M*) **using** *integrable* **by** (*simp add*: *nn-integral-eq-integral*)
also have ... ≤ *C* + |*a*| ∗ *emeasure M* (*space M*) **using** *assms*(*2*) *ennreal-leI* **by** *simp*
finally have ($\int^{+}\omega$. *max 0* (*X n* ω − *a*) ∂M) ≤ *C* + |*a*| ∗ *enn2real* (*emeasure M* (*space M*)) **using** *finite-emeasure-space C-nonneg* **by** (*simp add*: *ennreal-enn2real-if ennreal-mult*)
}
hence (*SUP N*. $\int^{+}$ *x*. *ennreal* (*max 0* (*X N x* − *a*)) ∂M) / (*b* − *a*) ≤ *ennreal* (*C* + |*a*| ∗ *enn2real* (*emeasure M* (*space M*))) / (*b* − *a*) **by** (*fast intro*: *divide-right-mono-ennreal Sup-least*)
moreover have *ennreal* (*C* + |*a*| ∗ *enn2real* (*emeasure M* (*space M*))) / (*b* − *a*) < ∞ **using** *that C-nonneg* **by** (*subst divide-ennreal*) *auto*
moreover have $integral^{N}$ *M* (*upcrossings X a b*) ≤ (*SUP N*. $\int^{+}$ *x*. *ennreal* (*max 0* (*X N x* − *a*)) ∂M) / (*b* − *a*)
using *upcrossing-inequality-Sup*[*OF assms*(*1*), *of b a*, *THEN divide-right-mono-ennreal*, *of b* − *a*]
ennreal-mult-divide-eq mult.commute[*of ennreal* (*b* − *a*)] *that* **by** *simp*
ultimately show *?thesis* **using** *upcrossings-measurable adapted-process-axioms that* **by** (*intro nn-integral-noteq-infinite*) *auto*
qed

— Since the number of upcrossings are finite, limsup and liminf have to agree almost everywhere. To show this we consider the following countable set, which has zero measure.

define *S* **where** *S* = ((λ(*a* :: *real*, *b*). {ω ∈ *space M*. *liminf* (λ*n*. *ereal* (*X n* ω)) < *ereal a* ∧ *ereal b* < *limsup* (λ*n*. *ereal* (*X n* ω))}) ' {(*a*, *b*) ∈ ℚ × ℚ. *a* < *b*})

have (*0*, *1*) ∈ {(*a* :: *real*, *b*). (*a*, *b*) ∈ ℚ × ℚ ∧ *a* < *b*} **unfolding** *Rats-def* **by** *simp*
moreover have *countable* {(*a*, *b*). (*a*, *b*) ∈ ℚ × ℚ ∧ *a* < *b*} **by** (*blast intro*: *countable-subset*[*OF - countable-SIGMA*[*OF countable-rat countable-rat*]])
ultimately have *from-nat-into-S*: *range* (*from-nat-into S*) = *S from-nat-into S n* ∈ *S* **for** *n*
unfolding *S-def*
by (*auto intro*!: *range-from-nat-into from-nat-into simp only*: *Rats-def*)
{
fix *a b* :: *real*
assume *a-less-b*: *a* < *b*
then obtain *N* **where** *N*: *x* ∈ *space M* − *N* ⟹ *upcrossings X a b x* ≠ ∞ *N* ∈ *null-sets M* **for** *x* **using** *AE-E3*[*OF finite-upcrossings*] **by** *blast*
{
fix ω
assume *liminf-limsup*: *liminf* (λ*n*. *X n* ω) < *a b* < *limsup* (λ*n*. *X n* ω)
have *upcrossings X a b* ω = ∞
proof −
{
fix *n*

have $\exists$ *m. upcrossings-before X a b m ω ≥ n*
proof (*induction n*)
case *0*
have *Sup* {*n. upcrossing X a b 0 n ω < 0*} = *0* **by** *simp*
then show *?case* **unfolding** *upcrossings-before-def* **by** *blast*
next
case (*Suc n*)
then obtain *m* **where** *m*: *n ≤ upcrossings-before X a b m ω* **by** *blast*
obtain *l* **where** *l*: *l ≥ m X l ω < a* **using** *liminf-upper-bound*[*OF liminf-limsup*(*1*), *of m*] *nless-le* **by** *auto*
obtain *u* **where** *u*: *u ≥ l X u ω > b* **using** *limsup-lower-bound*[*OF liminf-limsup*(*2*), *of l*] *nless-le* **by** *auto*
show *?case* **using** *upcrossings-before-less-exists-upcrossing*[*OF a-less-b*, **where** *?X=X*, *OF l u*] *m* **by** (*metis Suc-leI le-neq-implies-less*)
qed
}
thus *?thesis* **unfolding** *upcrossings-def* **by** (*simp add*: *ennreal-SUP-eq-top*)
qed
}
hence {*ω ∈ space M. liminf* (*λn. ereal* (*X n ω*)) < *ereal a* ∧ *ereal b* < *limsup* (*λn. ereal* (*X n ω*))} ⊆ *N* **using** *N* **by** *blast*
moreover have {*ω ∈ space M. liminf* (*λn. ereal* (*X n ω*)) < *ereal a* ∧ *ereal b* < *limsup* (*λn. ereal* (*X n ω*))} ∩ *N* ∈ *null-sets M* **by** (*force intro*: *null-set-Int1*[*OF N*(*2*)])
ultimately have *emeasure M* {*ω ∈ space M. liminf* (*λn. ereal* (*X n ω*)) < *a* ∧ *b* < *limsup* (*λn. ereal* (*X n ω*))} = *0* **by** (*simp add*: *Int-absorb1 Int-commute null-setsD1*)
}
hence *emeasure M* (*from-nat-into S n*) = *0* **for** *n* **using** *from-nat-into-S*(*2*)[*of n*] **unfolding** *S-def* **by** *force*
moreover have *S* ⊆ *M* **unfolding** *S-def* **by** *force*
ultimately have *emeasure M* (⋃ (*range* (*from-nat-into S*))) = *0* **using** *from-nat-into-S* **by** (*intro emeasure-UN-eq-0*) *auto*
moreover have (⋃ *S*) = {*ω ∈ space M. liminf* (*λn. ereal* (*X n ω*)) ≠ *limsup* (*λn. ereal* (*X n ω*))} (**is** *?L* = *?R*)
proof −
{
fix *ω*
assume *asm*: *ω ∈ ?L*
then obtain *a b* :: *real* **where** *a* < *b liminf* (*λn. ereal* (*X n ω*)) < *ereal a* ∧ *ereal b* < *limsup* (*λn. ereal* (*X n ω*)) **unfolding** *S-def* **by** *blast*
hence *liminf* (*λn. ereal* (*X n ω*)) ≠ *limsup* (*λn. ereal* (*X n ω*)) **using** *ereal-less-le order.asym* **by** *fastforce*
hence *ω ∈ ?R* **using** *asm* **unfolding** *S-def* **by** *blast*
}
moreover
{
fix *ω*
assume *asm*: *ω ∈ ?R*

hence *liminf* ($\lambda n.$ *ereal* (X n ω)) $<$ *limsup* ($\lambda n.$ *ereal* (X n ω)) **using** *Liminf-le-Limsup*[*of sequentially*] *less-eq-ereal-def* **by** *auto*
then obtain a' **where** a': *liminf* ($\lambda n.$ *ereal* (X n ω)) $<$ *ereal* a' *ereal* a' $<$ *limsup* ($\lambda n.$ *ereal* (X n ω)) **using** *ereal-dense2* **by** *blast*
then obtain b' **where** b': *ereal* a' $<$ *ereal* b' *ereal* b' $<$ *limsup* ($\lambda n.$ *ereal* (X n ω)) **using** *ereal-dense2* **by** *blast*
hence $a' < b'$ **by** *simp*
then obtain a **where** a: $a \in \mathbb{Q}$ $a' < a$ $a < b'$ **using** *Rats-dense-in-real* **by** *blast*
then obtain b **where** b: $b \in \mathbb{Q}$ $a < b$ $b < b'$ **using** *Rats-dense-in-real* **by** *blast*
have *liminf* ($\lambda n.$ *ereal* (X n ω)) $<$ *ereal* a **using** a a' *le-ereal-less order-less-imp-le* **by** *meson*
moreover have *ereal* b $<$ *limsup* ($\lambda n.$ *ereal* (X n ω)) **using** b b' *order-less-imp-le ereal-less-le* **by** *meson*
ultimately have $\omega \in$ *?L* **unfolding** *S-def* **using** a b *asm* **by** *blast*
}
ultimately show *?thesis* **by** *blast*
qed
ultimately have *emeasure* M $\{\omega \in$ *space* $M.$ *liminf* ($\lambda n.$ *ereal* (X n ω)) $\neq$ *limsup* ($\lambda n.$ *ereal* (X n ω))$\}$ $=$ *0* **using** *from-nat-into-S* **by** *argo*
hence *liminf-limsup-AE*: *AE* ω *in* $M.$ *liminf* ($\lambda n.$ X n ω) $=$ *limsup* ($\lambda n.$ X n ω) **by** (*intro AE-iff-measurable*[*THEN iffD2*, *OF - refl*]) *auto*
hence *convergent-AE*: *AE* ω *in* $M.$ *convergent* ($\lambda n.$ *ereal* (X n ω)) **using** *convergent-ereal* **by** *fastforce*

— Hence the limit exists almost everywhere.

have *bounded-pos-part*: *ennreal* ($\int \omega.$ *max 0* (X n ω) ∂M) $\leq$ *ennreal* C **for** n **using** *assms*(*2*) *ennreal-leI* **by** *blast*

— Integral of positive part is $< \infty$.

{
fix ω
assume *asm*: *convergent* ($\lambda n.$ *ereal* (X n ω))
hence ($\lambda n.$ *max 0* (*ereal* (X n ω))) $\longrightarrow$ *max 0* (*lim* ($\lambda n.$ *ereal* (X n ω)))
using *convergent-LIMSEQ-iff isCont-tendsto-compose continuous-max continuous-const continuous-ident continuous-at-e2ennreal*
by *fast*
hence ($\lambda n.$ *e2ennreal* (*max 0* (*ereal* (X n ω)))) $\longrightarrow$ *e2ennreal* (*max 0* (*lim* ($\lambda n.$ *ereal* (X n ω))))
using *isCont-tendsto-compose continuous-at-e2ennreal* **by** *blast*
moreover have *lim* ($\lambda n.$ *e2ennreal* (*max 0* (*ereal* (X n ω)))) $=$ *e2ennreal* (*max 0* (*lim* ($\lambda n.$ *ereal* (X n ω)))) **using** *limI calculation* **by** *blast*
ultimately have *e2ennreal* (*max 0* (*liminf* ($\lambda n.$ *ereal* (X n ω)))) $=$ *liminf* ($\lambda n.$ *e2ennreal* (*max 0* (*ereal* (X n ω)))) **using** *convergent-liminf-cl* **by** (*metis asm convergent-def limI*)
}
hence ($\int^{+}\omega.$ *e2ennreal* (*max 0* (*liminf* ($\lambda n.$ *ereal* (X n ω)))) ∂M) $=$ ($\int^{+}\omega.$

liminf (λn. *e2ennreal* (*max 0* (*ereal* (X n ω)))) ∂M) **using** *convergent-AE* **by** (*fast intro*: *nn-integral-cong-AE*)
moreover have ($\int^{+}\omega$. *liminf* (λn. *e2ennreal* (*max 0* (*ereal* (X n ω)))) ∂M) $\leq$ *liminf* (λn. ($\int^{+}\omega$. *e2ennreal* (*max 0* (*ereal* (X n ω))) ∂M))
by (*intro nn-integral-liminf*) *auto*
moreover have ($\int^{+}\omega$. *e2ennreal* (*max 0* (*ereal* (X n ω))) ∂M) $=$ *ennreal* ($\int \omega$. *max 0* (X n ω) ∂M) **for** n
using *e2ennreal-ereal ereal-max-0*
by (*subst nn-integral-eq-integral*[*symmetric*]) (*fastforce intro*!: *nn-integral-cong integrable* | *presburger*)+
moreover have *liminf-pos-part-finite*: *liminf* (λn. *ennreal* ($\int \omega$. *max 0* (X n ω) ∂M)) $< \infty$
unfolding *liminf-SUP-INF*
using *Inf-lower2*[*OF - bounded-pos-part*]
by (*intro order.strict-trans1*[*OF Sup-least*, *of - ennreal C*]) (*metis* (*mono-tags*, *lifting*) *atLeast-iff imageE image-eqI order.refl*, *simp*)
ultimately have *pos-part-finite*: ($\int^{+}\omega$. *e2ennreal* (*max 0* (*liminf* (λn. *ereal* (X n ω)))) ∂M) $< \infty$ **by** *force*

— Integral of negative part is $< \infty$.

{
fix ω
assume *asm*: *convergent* (λn. *ereal* (X n ω))
hence (λn. $-$ *min 0* (*ereal* (X n ω))) $\longrightarrow$ $-$ *min 0* (*lim* (λn. *ereal* (X n ω)))
using *convergent-LIMSEQ-iff isCont-tendsto-compose continuous-min continuous-const continuous-ident continuous-at-e2ennreal*
by *fast*
hence (λn. *e2ennreal* ($-$ *min 0* (*ereal* (X n ω)))) $\longrightarrow$ *e2ennreal* ($-$ *min 0* (*lim* (λn. *ereal* (X n ω))))
using *isCont-tendsto-compose continuous-at-e2ennreal* **by** *blast*
moreover have *lim* (λn. *e2ennreal* ($-$ *min 0* (*ereal* (X n ω)))) $=$ *e2ennreal* ($-$ *min 0* (*lim* (λn. *ereal* (X n ω)))) **using** *limI calculation* **by** *blast*
ultimately have *e2ennreal* ($-$ *min 0* (*liminf* (λn. *ereal* (X n ω)))) $=$ *liminf* (λn. *e2ennreal* ($-$ *min 0* (*ereal* (X n ω)))) **using** *convergent-liminf-cl* **by** (*metis asm convergent-def limI*)
}
hence ($\int^{+}\omega$. *e2ennreal* ($-$ *min 0* (*liminf* (λn. *ereal* (X n ω)))) ∂M) $=$ ($\int^{+}\omega$. *liminf* (λn. *e2ennreal* ($-$ *min 0* (*ereal* (X n ω)))) ∂M) **using** *convergent-AE* **by** (*fast intro*: *nn-integral-cong-AE*)
moreover have ($\int^{+}\omega$. *liminf* (λn. *e2ennreal* ($-$ *min 0* (*ereal* (X n ω)))) ∂M) $\leq$ *liminf* (λn. ($\int^{+}\omega$. *e2ennreal* ($-$ *min 0* (*ereal* (X n ω))) ∂M))
by (*intro nn-integral-liminf*) *auto*
moreover have ($\int^{+}\omega$. *e2ennreal* ($-$ *min 0* (*ereal* (X n ω))) ∂M) $=$ ($\int \omega$. *max 0* (X n ω) ∂M) $-$ ($\int \omega$. X n ω ∂M) **for** n
proof $-$
have $*$: ($-$ *min 0* c) $=$ *max 0* c $-$ c **if** $c \neq \infty$ **for** c :: *ereal* **using** *that* **by** (*cases* $c \geq 0$) *auto*

hence ($\int^+\omega$. *e2ennreal* (− *min 0* (*ereal* (*X n* ω))) ∂M) = ($\int^+\omega$. *e2ennreal* (*max 0* (*ereal* (*X n* ω)) − (*ereal* (*X n* ω))) ∂M) **by** *simp*
also have ... = ($\int^+ \omega$. *ennreal* (*max 0* (*X n* ω) − (*X n* ω)) ∂M) **using** *e2ennreal-ereal ereal-max-0 ereal-minus*(*1*) **by** (*intro nn-integral-cong*) *presburger*
also have ... = ($\int \omega$. *max 0* (*X n* ω) − (*X n* ω) ∂M) **using** *integrable* **by** (*intro nn-integral-eq-integral*) *auto*
finally show *?thesis* **using** *Bochner-Integration.integral-diff integrable* **by** *simp*
qed
moreover have *liminf* (λn. *ennreal* (($\int \omega$. *max 0* (*X n* ω) ∂M) − ($\int \omega$. *X n* ω ∂M))) < ∞
proof −
{
fix *n A*
assume *asm*: *ennreal* (($\int \omega$. *max 0* (*X n* ω) ∂M) − ($\int \omega$. *X n* ω ∂M)) $\in$ *A*
have ($\int \omega$. *X 0* ω ∂M) $\le$ ($\int \omega$. *X n* ω ∂M) **using** *set-integral-le*[*OF sets.top order-refl*, *of n*] *space-F* **by** (*simp add*: *integrable set-integral-space*)
hence ($\int \omega$. *max 0* (*X n* ω) ∂M) − ($\int \omega$. *X n* ω ∂M) $\le$ *C* − ($\int \omega$. *X 0* ω ∂M) **using** *assms*(*2*)[*of n*] **by** *argo*
hence *ennreal* (($\int \omega$. *max 0* (*X n* ω) ∂M) − ($\int \omega$. *X n* ω ∂M)) $\le$ *ennreal* (*C* − ($\int \omega$. *X 0* ω ∂M)) **using** *ennreal-leI* **by** *blast*
hence *Inf A* $\le$ *ennreal* (*C* − ($\int \omega$. *X 0* ω ∂M)) **by** (*rule Inf-lower2*[*OF asm*])
}
thus *?thesis*
unfolding *liminf-SUP-INF*
by (*intro order.strict-trans1*[*OF Sup-least*, *of - ennreal* (*C* − ($\int \omega$. *X 0* ω ∂M))]) (*metis* (*no-types*, *lifting*) *atLeast-iff imageE image-eqI order.refl order-trans*, *simp*)
qed
ultimately have *neg-part-finite*: ($\int^+ \omega$. *e2ennreal* (− (*min 0* (*liminf* (λn. *ereal* (*X n* ω))))) ∂M) < ∞ **by** *simp*

— Putting it all together now to show that the limit is integrable and < ∞ a.e.

have *e2ennreal* |*liminf* (λn. *ereal* (*X n* ω))| = *e2ennreal* (*max 0* (*liminf* (λn. *ereal* (*X n* ω)))) + *e2ennreal* (− (*min 0* (*liminf* (λn. *ereal* (*X n* ω))))) **for** ω
unfolding *ereal-abs-max-min*
by (*simp add*: *eq-onp-same-args max-def plus-ennreal.abs-eq*)
hence ($\int^+ \omega$. *e2ennreal* |*liminf* (λn. *ereal* (*X n* ω))| ∂M) = ($\int^+ \omega$. *e2ennreal* (*max 0* (*liminf* (λn. *ereal* (*X n* ω)))) ∂M) + ($\int^+ \omega$. *e2ennreal* (− (*min 0* (*liminf* (λn. *ereal* (*X n* ω))))) ∂M) **by** (*auto intro*: *nn-integral-add*)
hence *nn-integral-finite*: ($\int^+ \omega$. *e2ennreal* |*liminf* (λn. *ereal* (*X n* ω))| ∂M) $\neq$ ∞ **using** *pos-part-finite neg-part-finite* **by** *auto*
hence *finite-AE*: *AE* ω *in M*. *e2ennreal* |*liminf* (λn. *ereal* (*X n* ω))| $\neq$ ∞ **by** (*intro nn-integral-noteq-infinite*) *auto*
moreover
{
fix ω
assume *asm*: *liminf* (λn. *X n* ω) = *limsup* (λn. *X n* ω) |*liminf* (λn. *ereal* (*X n* ω))| $\neq$ ∞

hence $(\lambda n.\ X\ n\ \omega) \longrightarrow$ *real-of-ereal* (*liminf* $(\lambda n.\ X\ n\ \omega)$) **using** *limsup-le-liminf-real ereal-real′* **by** *simp*
}
ultimately have *converges*: *AE* ω *in* M. $(\lambda n.\ X\ n\ \omega) \longrightarrow$ *real-of-ereal* (*liminf* $(\lambda n.\ X\ n\ \omega)$) **using** *liminf-limsup-AE* **by** *fastforce*

{
fix ω
assume *e2ennreal* |*liminf* $(\lambda n.$ *ereal* $(X\ n\ \omega))| \neq \infty$
hence |*liminf* $(\lambda n.$ *ereal* $(X\ n\ \omega))| \neq \infty$ **by** *force*
hence *e2ennreal* |*liminf* $(\lambda n.$ *ereal* $(X\ n\ \omega))|$ = *ennreal* (*norm* (*real-of-ereal* (*liminf* $(\lambda n.$ *ereal* $(X\ n\ \omega)))))$ **by** *fastforce*
}
hence $(\int^{+} \omega.$ *e2ennreal* |*liminf* $(\lambda n.$ *ereal* $(X\ n\ \omega))|\ \partial M) = (\int^{+} \omega.$ *ennreal* (*norm* (*real-of-ereal* (*liminf* $(\lambda n.$ *ereal* $(X\ n\ \omega)))))\ \partial M)$ **using** *finite-AE* **by** (*fast intro*: *nn-integral-cong-AE*)
hence $(\int^{+} \omega.$ *ennreal* (*norm* (*real-of-ereal* (*liminf* $(\lambda n.$ *ereal* $(X\ n\ \omega)))))\ \partial M) < \infty$ **using** *nn-integral-finite* **by** (*simp add*: *order-less-le*)
hence *integrable* M $(\lambda\omega.$ *real-of-ereal* (*liminf* $(\lambda n.\ X\ n\ \omega)))$ **by** (*intro integrableI-bounded*) *auto*
moreover have $(\lambda\omega.$ *real-of-ereal* (*liminf* $(\lambda n.\ X\ n\ \omega))) \in$ *borel-measurable* F_∞ **using** *borel-measurable-liminf*[*OF F-infinity-measurableI*] *adapted* **by** *measurable*
ultimately show *?thesis* **using** *that converges* **by** *presburger*
qed

— We state the theorem again for martingales and supermartingales.

corollary *supermartingale-convergence-AE*:
fixes X :: *nat* $\Rightarrow$ *′a* $\Rightarrow$ *real*
assumes *supermartingale* $M\ F\ 0\ X$
and $\bigwedge n.\ (\int \omega.\ max\ 0\ (-\ X\ n\ \omega)\ \partial M) \leq C$
obtains X_{lim} **where** *AE* ω *in* M. $(\lambda n.\ X\ n\ \omega) \longrightarrow X_{lim}\ \omega$
integrable $M\ X_{lim}$
$X_{lim} \in$ *borel-measurable* (F_∞)
proof −
obtain Y **where** ∗: *AE* ω *in* M. $(\lambda n.\ -\ X\ n\ \omega) \longrightarrow Y\ \omega$ *integrable* $M\ Y\ Y \in$ *borel-measurable* (F_∞)
using *supermartingale.uminus*[*OF assms*(*1*), *THEN submartingale-convergence-AE*] *assms*(*2*) **by** *auto*
hence *AE* ω *in* M. $(\lambda n.\ X\ n\ \omega) \longrightarrow (-\ Y)\ \omega$ *integrable* $M\ (-\ Y)\ -\ Y \in$ *borel-measurable* (F_∞)
using *isCont-tendsto-compose*[*OF isCont-minus*, *OF continuous-ident*] *integrable-minus borel-measurable-uminus* **unfolding** *fun-Compl-def* **by** *fastforce*+
thus *?thesis* **using** *that*[*of* − Y] **by** *blast*
qed

corollary *martingale-convergence-AE*:
fixes X :: *nat* $\Rightarrow$ *′a* $\Rightarrow$ *real*
assumes *martingale* $M\ F\ 0\ X$

and $\bigwedge n.\ (\int \omega.\ |X\ n\ \omega|\ \partial M) \leq C$
obtains X_{lim} **where** $AE\ \omega\ in\ M.\ (\lambda n.\ X\ n\ \omega) \longrightarrow X_{lim}\ \omega$
integrable $M\ X_{lim}$
$X_{lim} \in$ *borel-measurable* (F_∞)
proof –
interpret *martingale-linorder* $M\ F\ 0\ X$ **unfolding** *martingale-linorder-def* **by** (*rule assms*)
have $max\ 0\ (X\ n\ \omega) \leq |X\ n\ \omega|$ **for** $n\ \omega$ **by** *linarith*
hence $(\int \omega.\ max\ 0\ (X\ n\ \omega)\ \partial M) \leq C$ **for** n **using** *assms*(*2*)[*THEN dual-order.trans*, *OF integral-mono*, *OF integrable-max*] *integrable* **by** *fast*
thus *?thesis* **using** *that submartingale-convergence-AE*[*OF submartingale-axioms*] **by** *blast*
qed

corollary *martingale-nonneg-convergence-AE*:
fixes $X :: nat \Rightarrow 'a \Rightarrow real$
assumes *martingale* $M\ F\ 0\ X$ $\bigwedge n.\ AE\ \omega\ in\ M.\ X\ n\ \omega \geq 0$
obtains X_{lim} **where** $AE\ \omega\ in\ M.\ (\lambda n.\ X\ n\ \omega) \longrightarrow X_{lim}\ \omega$
integrable $M\ X_{lim}$
$X_{lim} \in$ *borel-measurable* (F_∞)
proof –
interpret *martingale-linorder* $M\ F\ 0\ X$ **unfolding** *martingale-linorder-def* **by** (*rule assms*)
have $AE\ \omega\ in\ M.\ max\ 0\ (-\ X\ n\ \omega) = 0$ **for** n **using** *assms*(*2*)[*of n*] **by** *force*
hence $(\int \omega.\ max\ 0\ (-\ X\ n\ \omega)\ \partial M) \leq 0$ **for** n **by** (*simp add*: *integral-eq-zero-AE*)
thus *?thesis* **using** *that supermartingale-convergence-AE*[*OF supermartingale-axioms*] **by** *blast*
qed

end

end

References

[1] R. Degenne and K. Ying. A Formalization of Doob's Martingale Convergence Theorems in mathlib. In *Proceedings of the 12th ACM SIGPLAN International Conference on Certified Programs and Proofs*. Association for Computing Machinery, New York, United States, 2022. arXiv:2212.05578.

[2] R. Degenne and K. Ying. probability.martingale.basic - mathlib, 2022. https://leanprover-community.github.io/mathlib_docs/probability/martingale/basic.html, Last Accessed: 15 Feb 2024.

[3] R. Durrett. *Probability: Theory and Examples.* Cambridge Series in Statistical and Probabilistic Mathematics. Cambridge University Press, 2019. URL: https://books.google.de/books?id=b22MDwAAQBAJ.

[4] G. Grimmett and D. Stirzaker. *Probability and Random Processes*. Oxford University Press, 2020.

[5] J. Hölzl and A. Heller. Three Chapters of Measure Theory in Isabelle/HOL. In M. C. J. D. van Eekelen, H. Geuvers, J. Schmaltz, and F. Wiedijk, editors, *Interactive Theorem Proving (ITP 2011)*, volume 6898 of *LNCS*, pages 135–151, 2011. doi:10.1007/978-3-642-22863-6_12.

[6] A. Keskin. A Formalization of Martingales in Isabelle/HOL. Bachelor's thesis, Technical University of Munich, 2023. arXiv:2311.06188.

[7] A. Keskin. Martingales. *Archive of Formal Proofs*, November 2023. https://isa-afp.org/entries/Martingales.html, Formal proof development.